

Resolutions and Cohomology over Complete Intersections

Luichezar L. Avramov and Daniel R. Grayson*

This chapter contains a new proof and new applications of a theorem of Shamash and Eisenbud, providing a construction of projective resolutions of modules over a complete intersection. The duals of these infinite projective resolutions are finitely generated differential graded modules over a graded polynomial ring, so they can be represented in the computer, and can be used to compute Ext modules simultaneously in all homological degrees. It is shown how to write *Macaulay 2* code to implement the construction, and how to use the computer to determine invariants of modules over complete intersections that are difficult to obtain otherwise.

Introduction

Let $A = K[x_1, \dots, x_e]$ be a polynomial ring with variables of positive degree over a field K , and $B = A/J$ a quotient ring modulo a homogeneous ideal.

In this paper we consider the case when B is a *graded complete intersection*, that is, when the defining ideal J is generated by a homogeneous A -regular sequence. We set up, describe, and illustrate a routine `Ext`, now implemented in *Macaulay 2*. For any two finitely generated graded B -modules M and N it yields a presentation of $\text{Ext}_B^\bullet(M, N)$ as a bigraded module over an appropriately bigraded polynomial ring $S = A[X_1, \dots, X_c]$.

A novel feature of our routine is that it computes the modules $\text{Ext}_B^n(M, N)$ *simultaneously in all cohomological degrees* $n \geq 0$. This is made possible by the use of *cohomology operations*, a technique usually confined to theoretical considerations. Another aspect worth noticing is that, although the result is over a ring B with nontrivial relations, all the computations are made over the *polynomial ring* S ; this may account for the effectiveness of the algorithm.

To explain the role of the complete intersection hypothesis, we cast it into the broader context of homological algebra over graded rings.

Numerous results indicate that the high syzygy modules of M exhibit ‘similar’ properties. For an outrageous example, assume that M has finite projective dimension. Its distant syzygies are then all equal to 0, and so—for trivial reasons—display an extremely uniform behavior. However, even this case has a highly nontrivial aspect: due to the Auslander-Buchsbaum Equality asymptotic information is available after at most $(e + 1)$ steps. This accounts for the effectiveness of computer constructions of *finite* free resolutions.

* Authors supported by the NSF, grants DMS 99-70375 and DMS 99-70085.

Problems that computers are not well equipped to handle arise unavoidably when studying asymptotic behavior of *infinite* resolutions. We describe some, using graded Betti numbers $\beta_{ns}^B(M) = \dim_K \text{Ext}_B^n(M, k)_{-s}$, where $k = B/(x_1, \dots, x_e)B$, and regularity $\text{reg}_B(M) = \sup_{n,s} \{s - n \mid \beta_{ns}^B(M) \neq 0\}$.

- *Irrationality.* There are rings B for which no recurrent relation with constant coefficients exists among the numbers $\beta_n^B(k) = \sum_s \beta_{ns}^B(k)$, see [1].
- *Irregularity.* For each $r \geq 2$ there exists a ring $B(r)$ with $\beta_{ns}^{B(r)}(k) = 0$ for $s \neq n$ and $0 \leq n \leq r$, but with $\beta_{r,r+1}^{B(r)}(k) \neq 0$, see [14].
- *Span.* If B is generated over K by elements of degree one and $\text{reg}_B(k) \neq 0$, then $\text{reg}_B(k) = \infty$, see [7].
- *Size.* There are inequalities $\beta_n^B(k) \geq \beta^n$ for all $n \geq 0$ and for some constant $\beta > 1$, unless B is a complete intersection, see [3].

These obstructions vanish miraculously when B is a graded complete intersection: For each M and all $n \gg 0$ the number $\beta_{n+1}(M)$ is a linear combination with constant coefficients of $\beta_{n-2c}^B(M), \dots, \beta_n^B(M)$. If B is generated in degree one, then $\text{reg}_B(k) = 0$ if and only if the ideal J is generated by quadratic forms. There are inequalities $\beta_n^B(M) \leq \beta(M)n^{c-1}$ for all $n \geq 1$ and for some constant $\beta(M) > 0$.

The algebra behind the miracle is a theorem of Gulliksen [12], who proves that $\text{Ext}_B^*(M, N)$ is a finitely generated bigraded module over a polynomial ring of cohomology operators $S = A[X_1, \dots, X_c]$, where each variable X_i has cohomological degree 2. As a consequence of this result, problems in Homological Algebra can be answered in terms of Commutative Algebra.

Gulliksen's definition of the operators X_i as iterated connecting homomorphisms is badly suited for use by a computer. Other definitions have been given subsequently by several authors, see Remark 4.6. We take the approach of Eisenbud [11], who derives the operators from a specific B -free resolution of M , obtained by extending a construction of Shamash [15].

The resolution of Shamash and Eisenbud, and Gulliksen's Finiteness Theorem, are presented with detailed proofs in Section 4. They are obtained through a new construction—that of an intermediate resolution of M over the polynomial ring—that encodes C and all the null-homotopies of C corresponding to multiplication with elements of J ; this material is contained in Section 3. It needs standard multilinear algebra, developed *ad hoc* in Section 2. Rules for juggling several gradings are discussed in an Appendix.

In Section 5 we present and illustrate the code for the routine `Ext`, which runs remarkably close to the proofs in Sections 3 and 4. Section 6 contains numerous computations of popular numerical invariants of a graded module, like its complexity, Poincaré series, and Bass series. They are extracted from knowledge of the bigraded modules $\text{Ext}_B^*(M, k)$ and $\text{Ext}_B^*(k, M)$, whose computation is also illustrated by examples, and is further used to obtain explicit equations for the cohomology variety $V_B^*(M)$ defined in [2]. For most invariants we include some short code that automates their computation. In Section 7 we extend these procedures to invariants of pairs of modules.

1 Matrix Factorizations

We start the discussion of homological algebra over a complete intersection with a very special case, that can be packaged attractively in matrix terms.

Let f be a non-zero-divisor in a commutative ring A .

Following Eisenbud [11, Sect. 5] we say that a pair (U, V) of matrices with entries in A , of sizes $k \times \ell$ and $\ell \times k$, is a *matrix factorization* of $-f$ if

$$U \cdot V = -f \cdot I_k \quad \text{and} \quad V \cdot U = -f \cdot I_\ell$$

where I_m denotes the $m \times m$ unit matrix. Localizing at f , one sees that $-f^{-1} \cdot U$ and V are inverse matrices over A_f ; as a consequence $\ell = k$, and each equality above implies the other, for instance:

$$V \cdot U = (-f^{-1} \cdot U)^{-1} \cdot U = -f \cdot U^{-1} \cdot U = -f \cdot I_k$$

Here is a familiar example of matrix factorization, with $f = xy - wz$:

$$\begin{pmatrix} w & x \\ y & z \end{pmatrix} \cdot \begin{pmatrix} z & -y \\ -x & w \end{pmatrix} = -(xy - wz) \cdot \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} z & -y \\ -x & w \end{pmatrix} \cdot \begin{pmatrix} w & x \\ y & z \end{pmatrix}$$

Let now C_1 and C_0 be free A -modules of rank r , and let

$$d_1: C_1 \rightarrow C_0 \quad \text{and} \quad s_0: C_0 \rightarrow C_1$$

be A -linear homomorphisms defined by the matrices U and V , respectively, after bases have been tacitly chosen.

The second condition on the matrices U and V implies that d_1 is injective, while the first condition on these matrices shows that fC_0 is contained in $\text{Im}(d_1)$. Setting $L = \text{Coker}(d_1)$, one sees that the chosen matrix factorization defines a commutative diagram with exact rows

$$\begin{array}{ccccccccc} 0 & \longrightarrow & C_1 & \xrightarrow{d_1} & C_0 & \longrightarrow & L & \longrightarrow & 0 \\ & & \downarrow -f \cdot 1_{C_1} & \swarrow s_0 & \downarrow -f \cdot 1_{C_0} & & \downarrow 0_L = -f \cdot 1_L & & \\ 0 & \longrightarrow & C_1 & \xrightarrow{d_1} & C_0 & \longrightarrow & L & \longrightarrow & 0 \end{array}$$

which expresses the following facts: $C = 0 \rightarrow C_1 \xrightarrow{d_1} C_0 \rightarrow 0$ is a free resolution of the A -module L , this module is annihilated by f , and s_0 is a homotopy between the maps $-f \cdot 1_C$ and 0_C , both of which lift $-f \cdot 1_L$.

Conversely, if an A -module L annihilated by f has a free resolution (C, d_1) of length 1, then $\text{rank}_A C_1 = \text{rank}_A C_0$, and any choice of homotopy s_0 between $-f \cdot 1_C$ and 0_C provides a matrix factorization of $-f$.

When we already have an A -module L with a presentation matrix U that defines an injective A -linear map, we can use *Macaulay 2* to create a matrix factorization (U, V) of $-f$.

Example 1.1. We revisit the familiar example from a higher perspective.

```

i1 : A = QQ[w,x,y,z]
o1 = A
o1 : PolynomialRing
i2 : U = matrix {{w,x},{y,z}}
o2 = | w x |
      | y z |
o2 : Matrix A <--- A
i3 : C = chainComplex U
o3 = A <--- A
      0 1
o3 : ChainComplex
i4 : L = HH_0 C
o4 = cokernel | w x |
              | y z |
o4 : A-module, quotient of A
i5 : f = -det U
o5 = x*y - w*z
o5 : A

```

Let's verify that f annihilates L .

```

i6 : f * L == 0
o6 = true

```

We use the `nullhomotopy` function.

```

i7 : s = nullhomotopy (-f * id_C)
o7 = 1 : A <----- A : 0
      {1} | z -x |
      {1} | -y w |
o7 : ChainComplexMap

```

Let's verify that s is a null-homotopy for $-f$, using `C.dd` to obtain the differential of the chain complex C as a map of graded modules.

```

i8 : s * C.dd + C.dd * s == -f
o8 = true

```

We extract the matrix V from the null-homotopy to get our factorization.

```

i9 : V = s_0
o9 = {1} | z -x |
      {1} | -y w |

```

```

o9 : Matrix A <--- A

```

For every f and every $r \geq 1$ there exists a trivial matrix factorization of $-f$, namely, $(f \cdot I_k, -I_k)$; it can be obtained from the A -module $L = A^k/fA^k$. In general, it may not be clear how to find an A -module L with the properties necessary to obtain an ‘interesting’ matrix factorization of $-f$.

However, in some cases the supply is plentiful.

Remark 1.2. Let A be a graded polynomial ring in e variables of positive degree over a field K , let f be a homogeneous polynomial in A , and set $B = A/(f)$. Every B -module M of infinite projective dimension *generates* a matrix factorization (U, V) of $-f$, as follows.

Let (F, d_F) be a minimal graded free resolution of M over B , and set $L = \text{Coker}(d_F: F_{e+1} \rightarrow F_e)$. Since M has infinite projective dimension, we have $L \neq 0$. By the Depth Lemma, $\text{depth}_B L = \text{depth } B$. On the other hand, $\text{depth}_B L = \text{depth}_A L$ and $\text{depth } B = \text{depth } A - 1$. By Hilbert’s Syzygy Theorem, the minimal graded free resolution (C, d_C) of L over A is finite. By the Auslander-Buchsbaum Equality, $C_n = 0$ for $n > \text{depth } A - \text{depth}_A L = 1$.

The minimality of F ensures that all nonzero entries of the presentation matrix U of L are forms of positive degree. On the other hand, by [11, Sect. 0] the module L has no direct summand isomorphic to B : it follows that all nonzero entries of the homotopy matrix V are forms of positive degree (this is the reason for choosing L as above—stopping one step earlier in the resolution F could have produced a module L with a non-zero free direct summand).

Our reader would have noticed that *Macaulay 2* can read all the data and perform all the operations needed to construct a module L by the procedure described in the preceding remark. Here is how it does it.

Example 1.3. We produce a matrix factorization of $-f$, where

$$f = x^3 + 3y^3 - 2yz^2 + 5z^3 \in \mathbb{Q}[x, y, z] = A$$

generated by the module $M = B/\mathfrak{m}^2$, where $B = A/(f)$ and $\mathfrak{m} = (x, y, z)B$.

```

i10 : A = QQ[x,y,z];
i11 : f = x^3 + 3*y^3 - 2*y*z^2 + 5*z^3;
i12 : B = A/f;
i13 : m = ideal(x,y,z)
o13 = ideal (x, y, z)
o13 : Ideal of B

```

Let’s take the B -module M and compute its minimal B -free resolution.

```

i14 : M = B^1/m^2;

```

```

i15 : F = resolution(M, LengthLimit=>8)

      1      6      9      9      9      9      9      9      9
o15 = B <-- B <-- B <-- B <-- B <-- B <-- B <-- B <-- B
      0      1      2      3      4      5      6      7      8

o15 : ChainComplex

```

We introduce a function `restrict1 N` which accepts a B -module N and restricts scalars to produce an A -module.

```

i16 : restrict1 = N -> coker(lift(presentation N,A) | f);

```

Now make L as described in Remark 1.2

```

i17 : L = restrict1 cokernel F.dd_4;

```

We proceed as in Example 1.1 to get a matrix factorization.

```

i18 : C = res L;

i19 : U = C.dd_1;

      9      9
o19 : Matrix A <--- A

i20 : print U
{4} | 0 xy x2      y2  0      0      0      yz-5/2z2 0      |
{4} | 0 x2 -3y2    xy   yz-5/2z2 0      yz-5/2z2 0      0      |
{4} | x2 0 -2yz+5z2 0    y2-5/2yz yz-5/2z2 -5/2yz 0      0      |
{5} | 0 0 0      1/3z 0      0      0      1/2y    x      |
{5} | 0 0 -z      0    1/2y  0      1/2y    -1/2x   0      |
{5} | 0 -z 0      0    -1/2x 0      -1/2x   0      3y     |
{5} | 0 0 0      -1/3x 0      1/2y  -1/3z   0      0      |
{5} | -z y x      0    0      -1/2x 0      0      0      |
{5} | y 0 0      0    0      0      1/3x   0      -2y+5z |

i21 : s = nullhomotopy (-f * id_C);

i22 : V = s_0;

      9      9
o22 : Matrix A <--- A

i23 : print V
{6} | 0 0 -x 0      0      -2y2+5yz 0      -2yz+5z2 -3y2 |
{6} | 0 -x 0 0      0      -2yz+5z2 -3xy   -3y2    -3yz |
{6} | -x y 0 0      -2yz+5z2 0      0      0      0      |
{6} | -3y 0 0 6yz-15z2 0      0      3x2    3xy    3xz |
{6} | 0 2z -3y -15xz -15yz 2x2 6yz-15z2 0      3x2 |
{6} | -2x 0 2z 0      -4yz+10z2 0      -6y2    2x2    0      |
{6} | 0 0 3y -6xy+15xz -6y2+15yz 0      -6yz+15z2 0      -3x2 |
{6} | 2z 0 0 -6y2 2x2 2xy 0      0      0      |
{6} | 0 0 0 -x2 -xy -y2 -xz -yz -z2 |

i24 : U*V+f==0

o24 = true

i25 : V*U+f==0

o25 = true

```

The procedure described above can be automated for more pleasant usage.

Code 1.4. The function `matrixFactorization` `M` produces a matrix factorization (U, V) of $-f$ generated by a module M over $B = A/(f)$.

```
i26 : matrixFactorization = M -> (
      B := ring M;
      f := (ideal B)_0;
      e := numgens B;
      F := resolution(M, LengthLimit => e+1);
      L := restrict1 cokernel F.dd_(e+1);
      C := res L;
      U := C.dd_1;
      s := nullhomotopy (-f * id_C);
      V := s_0;
      assert( U*V + f == 0 );
      assert( V*U + f == 0 );
      return (U,V);
```

We use the `assert` command to signal an error in case the matrices found don't satisfy our requirement for a matrix factorization.

Let's illustrate the new code with a slightly bigger module M than before.

Example 1.5. With the same A , f , B , and \mathfrak{m} as in Example 1.3, we produce a matrix factorization generated by the B -module $M = B/\mathfrak{m}^3$.

```
i27 : time (U,V) = matrixFactorization(B^1/m^3);
      -- used 0.21 seconds
```

The parallel assignment statement above provides both variables `U` and `V` with matrix values. We examine their shapes without viewing the matrices themselves by appending a semicolon to the appropriate command.

```
i28 : U;
      15      15
o28 : Matrix A  <--- A
i29 : V;
      15      15
o29 : Matrix A  <--- A
```

Matrix factorizations were introduced to construct resolutions over the the residue ring $B = A/(f)$, using the following observation.

Remark 1.6. If (U, V) is a factorization of $-f$ by $k \times k$ matrices and the maps $d_1: C_1 \rightarrow C_0$ and $s_0: C_0 \rightarrow C_1$ are homomorphisms of free A -modules defined by U and V , respectively, then the sequence

$$\cdots \rightarrow C_1 \otimes_A B \xrightarrow{d_1 \otimes 1_B} C_0 \otimes_A B \xrightarrow{s_0 \otimes 1_B} C_1 \otimes_A B \xrightarrow{d_1 \otimes 1_B} C_0 \otimes_A B \rightarrow 0$$

of B -linear maps is a free resolution of the B -module $L = \text{Coker}(d_1)$.

Indeed, freeness is clear, and we have a complex because $d_1 s_0 = -f \cdot 1_{C_0}$ and $s_0 d_1 = -f \cdot 1_{C_1}$. If $x \in C_1$ satisfies $(d_1 \otimes 1_B)(x \otimes 1) = 0$, then $d_1(x) = fy$ for some $y \in C_0$, hence $d_1 x = d_1 s_0(y)$. As d_1 is injective, we get $x = s_0(y)$, so $\text{Ker}(d_1 \otimes 1_B) \subseteq \text{Im}(s_0 \otimes 1_B)$; the reverse inclusion follows by symmetry.

Pooling Remarks 1.2 and 1.6 we recover Eisenbud's result [11, Sect. 6].

Theorem 1.7. *Let A be a graded polynomial ring in e variables of positive degree over a field K , and f a homogeneous polynomial in A . The minimal graded free resolution of every finitely generated graded module over $B = A/(f)$ becomes periodic of period 2 after at most e steps. The periodic part of the resolution is given by a matrix factorization of $-f$ generated by M .*

We illustrate the theorem on an already computed example.

Example 1.8. Let A , f , B , M , and F be as in Example 1.3.

To verify the periodicity of F we subtract pairs of differentials and compare the result with 0: direct comparison of the differentials would not work, because the free modules involved have different degrees.

```
i30 : F.dd_3 - F.dd_5 == 0
o30 = false
i31 : F.dd_4 - F.dd_6 == 0
o31 = false
i32 : F.dd_5 - F.dd_7 == 0
o32 = true
```

The first two answers above come as a surprise—and suggest a property of F that is weaker than the one we already know to be true!

There is an easy explanation: we checked the syzygy modules for *equality*, rather than for *isomorphism*. We do not know why *Macaulay 2* didn't produce an equality at the earliest possible stage, nor why it eventually produced one. The program has other strategies for computing resolutions, so let's try one.

```
i33 : M = B^1/m^2;
i34 : G = resolution(M, LengthLimit => 8, Strategy => 0)
o34 = B  <-- B  <-- B  <-- B  <-- B  <-- B  <-- B  <-- B  <-- B
      0    1    2    3    4    5    6    7    8
o34 : ChainComplex
i35 : G.dd_3 - G.dd_5 == 0
o35 = true
i36 : G.dd_4 - G.dd_6 == 0
o36 = true
i37 : G.dd_5 - G.dd_7 == 0
o37 = true
```

The strategy paid off, revealing periodicity at the earliest possible stage. However, the algorithm used may be a lot slower than the default algorithm.

2 Graded Algebras

We describe some standard universal algebras over a commutative ring A .

Let Q denote a free A -module of rank c , and set $Q^* = \text{Hom}_A(Q, A)$. We assign degree 2 to the elements of Q , and degree -2 to those of Q^* . We let Q^\wedge denote a copy of Q whose elements are assigned degree 1; if x is an element of Q , then x^\wedge denotes the corresponding element of Q^\wedge .

We use $\alpha = (\alpha_1, \dots, \alpha_c) \in \mathbb{Z}^c$ as a multi-index, set $|\alpha| = \sum_i \alpha_i$, and order \mathbb{Z}^c by the rule: $\alpha \geq \beta$ means $\alpha_i \geq \beta_i$ for each i . We let o denote the trivial element of \mathbb{Z}^c , and ϵ_i the i 'th element of its standard basis.

Construction 2.1. For each integer $m \geq 0$ we form free A -modules

$$\begin{aligned} S^m(Q^*) & \text{ with basis } \{X^\alpha : |\alpha| = m\} \\ D^m(Q) & \text{ with basis } \{Y^{(\alpha)} : |\alpha| = m\} \\ \bigwedge^m(Q^\wedge) & \text{ with basis } \{Y^{\wedge\alpha} : |\alpha| = m \text{ and } \alpha \leq (\epsilon_1 + \dots + \epsilon_c)\} \end{aligned}$$

For $m < 0$ we declare the modules $S^m(Q^*)$, $D^m(Q)$, and $\bigwedge^m(Q^\wedge)$ to be equal to 0, and define the symbols X^α , $Y^{(\alpha)}$, and $Y^{\wedge\alpha}$ accordingly; in addition, we set $\bigwedge^m(Q^\wedge) = 0$ and $Y^{\wedge\alpha} = 0$ if $|\alpha| \not\leq (\epsilon_1 + \dots + \epsilon_c)$, and we set

$$X_i = X^{\epsilon_i} \quad Y_i = Y^{(\epsilon_i)} \quad Y_i^\wedge = Y^{\wedge\epsilon_i} \quad \text{for } i = 1, \dots, c$$

Taking $S^m(Q^*)$, $D^m(Q)$, and $\bigwedge^m(Q^\wedge)$ as homogeneous components of degree $-2m$, $2m$, and m , respectively, we introduce graded algebras

$$S = S(Q^*) \quad D = D(Q) \quad E = \bigwedge(Q^\wedge)$$

by defining products of basis elements by the formulas

$$\begin{aligned} X^\alpha \cdot X^\beta &= X^{\alpha+\beta} \\ Y^{(\alpha)} \cdot Y^{(\beta)} &= \prod_{i=1}^c \frac{(\alpha_i + \beta_i)!}{\alpha_i! \beta_i!} Y^{(\alpha+\beta)} \\ Y^{\wedge\alpha} \cdot Y^{\wedge\beta} &= \text{inv}(\alpha, \beta) Y^{\wedge\alpha+\beta} \end{aligned}$$

where $\text{inv}(\alpha, \beta)$ denotes the number of pairs (i, j) with $\alpha_i = \beta_j = 1$ and $i > j$. Thus, S is the *symmetric algebra* of Q^* , with $X^o = 1$, while D is the *divided powers algebra* of Q , with $Y^{(o)} = 1$, and E is the *exterior algebra* of Q^\wedge , with $Y^{\wedge o} = 1$. We identify S and the polynomial ring $A[X_1, \dots, X_c]$.

A *homogeneous derivation* of a graded A -algebra W is a homogeneous A -linear map $d: W \rightarrow W$ such that the *Leibniz rule*

$$d(xy) = d(x)y + (-1)^{\deg x \cdot \deg d} x d(y)$$

holds for all homogeneous elements $x, y \in W$.

Construction 2.2. Each sequence $f_1, \dots, f_c \in A$ yields a *Koszul map*

$$d_E: E \rightarrow E \quad \text{defined by the formula}$$

$$d_E(Y^{\wedge\beta}) = \sum_{i=1}^c (-1)^{\beta_1 + \dots + \beta_{i-1}} f_i Y^{\wedge\beta - \epsilon_i}$$

It is a derivation of degree -1 and satisfies $d_E^2 = 0$.

Construction 2.3. For every $X_i \in S^1(Q^*)$ and each $Y^{(\beta)} \in D^m(Q)$ we set

$$X_i \lrcorner Y^{(\beta)} = Y^{(\beta - \epsilon_i)} \in D^{m-1}(Q)$$

Extending this formula by A -bilinearity, we define $g \lrcorner y$ for all $g \in S^1(Q^*)$ and all $y \in D$. It is well known, and easily verified, that the map $g \lrcorner: y \mapsto g \lrcorner y$ is a graded derivation $D \rightarrow D$ of degree -2 , and that the derivations associated with arbitrary g and g' commute. As a consequence, the formula

$$X^\alpha \lrcorner Y^{(\beta)} = (X_1 \lrcorner)^{\alpha_1} \dots (X_c \lrcorner)^{\alpha_c} (Y^{(\beta)}) \in D^{|\beta - \alpha|}(Q)$$

extended A -linearly to all $u \in S$, defines on D a structure of graded S -module.

The usual products on $S \otimes_A E$ and $D \otimes_A E$ and the induced gradings

$$(S \otimes_A E)_n = \bigoplus_{\ell - 2k = n} S^k(Q^*) \otimes_A \wedge^\ell(Q^\wedge)$$

$$(D \otimes_A E)_n = \bigoplus_{\ell + 2k = n} D^{(k)}(Q) \otimes_A \wedge^\ell(Q^\wedge)$$

turn $S \otimes_A E$ and $D \otimes_A E$ into graded algebras. The second one is a graded module over the first, for the action $(u \otimes z) \cdot (y \otimes z') = (u \lrcorner y) \otimes (z \cdot z')$.

Construction 2.4. The element $w = \sum_{i=1}^c X_i \otimes Y_i^\wedge$ yields a *Cartan map*

$$d_{DE}: D \otimes_A E \rightarrow D \otimes_A E \quad \text{defined by the formula}$$

$$d_{DE}(y \otimes z) = w \cdot (y \otimes z) = \sum_{i=1}^c (X_i \lrcorner y) \otimes (Y_i^\wedge \cdot z)$$

It is an E -linear derivation of degree -1 , and $d_{DE}^2 = 0$ because $w^2 = 0$.

Lemma 2.5. For each integer s define a complex G^s as follows:

$$\dots \rightarrow D^k(Q) \otimes_A \wedge^{s-k}(Q^\wedge) \xrightarrow{w} D^{k-1}(Q) \otimes_A \wedge^{s-k+1}(Q^\wedge) \rightarrow \dots$$

with $D^0(Q) \otimes_A \wedge^s(Q^\wedge)$ in degree s . If $s > 0$, then G^s is split exact.

Proof. Note that for each $s \in \mathbb{Z}$ there exist isomorphisms of complexes $\bigoplus_{s=1}^\infty G^s \cong (D \otimes_A E)_{\geq 1} \cong \left(\bigotimes_{i=1}^c G(i) \right)_{\geq 1}$, where $G(i)$ is the complex

$$\dots \rightarrow AY_i^{(k+1)} \otimes_A A \xrightarrow{w_i} (AY_i^{(k)}) \otimes (AY_i^\wedge) \xrightarrow{0} (AY_i^{(k)}) \otimes_A A \rightarrow \dots$$

and w_i is left multiplication with $X_i \otimes Y_i^\wedge$. This map is bijective, so each complex $G(i)_{\geq 1}$ is split exact. The assertion follows. \square

3 Universal Homotopies

This section contains the main new mathematical result of the paper.

We introduce a universal construction, that takes as input a projective resolution C of an A -module M and a finite set \mathbf{f} of elements annihilating M ; the output is a new projective resolution of M over A . If $\mathbf{f} \neq \emptyset$, then the new resolution is infinite—even when C is finite—because it encodes additional data: the null-homotopies for $f \cdot 1_C$ for all $f \in \mathbf{f}$, all compositions of such homotopies, and all relations between those compositions. This higher-order information tracks the transformation of the homological properties of M when its ring of operators is changed from A to $A/(\mathbf{f})$.

Our construction is motivated by, and is similar to, one due to Shamash [15] and Eisenbud [11]: assuming that the elements of \mathbf{f} form an A -regular sequence, they produce a projective resolution of M over $A/(\mathbf{f})$. By contrast, we make no assumption whatsoever on \mathbf{f} . With the additional hypothesis, in the next section we quickly recover the original result from the one below. As an added benefit, we eliminate the use of spectral sequences from the proof.

Theorem 3.1. *Let A be a commutative ring, let f_1, \dots, f_c be a sequence of elements of A , let M be an A -module annihilated by f_i for $i = 1, \dots, c$, and let $r: C \rightarrow M$ be a resolution of M by projective (respectively, free) A -modules.*

There exists a family of homogeneous A -linear maps

$$\{d_\gamma: C \rightarrow C \mid \deg(d_\gamma) = 2|\gamma| - 1\}_{\gamma \in \mathbb{N}^c}$$

satisfying the following conditions

$$\begin{aligned} d_o &= d_C \quad \text{is the differential of } C \\ [d_o, d_\gamma] &= \begin{cases} -f_i \cdot 1_C & \text{if } \gamma = \epsilon_i \text{ for } i = 1, \dots, c \\ -\sum_{\alpha+\beta=\gamma}^+ d_\alpha d_\beta & \text{if } |\gamma| \geq 2 \end{cases} \end{aligned} \quad (1)$$

where \sum^+ denotes a summation restricted to indices in $\mathbb{N}^c \setminus \{o\}$.

Any family $\{d_\gamma\}_{\gamma \in \mathbb{N}^c}$ as above defines an A -linear map of degree -1 ,

$$\begin{aligned} d_{CD}: C \otimes_A D &\rightarrow C \otimes_A D \quad \text{given by} \\ d_{CD}(x \otimes y) &= \sum_{\gamma \in \mathbb{N}^c} d_\gamma(x) \otimes (X^\gamma \lrcorner y) \end{aligned} \quad (2)$$

where D is the divided powers algebra defined in Construction 2.1, and the action of X^γ on D is defined in Construction 2.3.

With d_E and d_{DE} defined in Constructions 2.2 and 2.4 and the tensor product of maps of graded modules defined as in Remark 3.4, the map

$$\begin{aligned} d: C \otimes_A D \otimes_A E &\rightarrow C \otimes_A D \otimes_A E \quad \text{given by} \\ d &= d_{CD} \otimes 1_E + 1_C \otimes d_{DE} + 1_C \otimes 1_D \otimes d_E \end{aligned} \quad (3)$$

is an A -linear differential of degree -1 , and the map

$$q: C \otimes_A D \otimes_A E \rightarrow M \quad \text{given by}$$

$$q(x \otimes y \otimes z) = \begin{cases} yz \cdot r(x) & \text{if } \deg(y) = \deg(z) = 0 \\ 0 & \text{otherwise} \end{cases}$$

is a resolution of M by projective (respectively, free) A -modules.

For use in the proof, we bring up a few general homological points.

A *bounded filtration* of a chain complex F is a sequence

$$0 = F^0 \subseteq F^1 \subseteq \dots \subseteq F^{s-1} \subseteq F^s \subseteq \dots$$

of subcomplexes such that for each n there exists an s with $F_n^s = F_n$. As usual, we let $\mathrm{gr}^s(F)$ denote the complex of A -modules F^s/F^{s-1} .

Lemma 3.2. *Let $q: F \rightarrow F'$ be a morphism of complexes with bounded filtrations, such that $q(F^s) \subseteq F'^s$ for all $s \in \mathbb{Z}$. If for each s the induced map $\mathrm{gr}^s(q): \mathrm{gr}^s(F) \rightarrow \mathrm{gr}^s(F')$ is a quasi-isomorphism, then so is q .*

Proof. Denoting q^s the restriction of q to F^s , we first show by induction on s that $H_n(q^s)$ is bijective for all n . The assertion is clear for $s = 0$, since $F^0 = 0$ and $F'^0 = 0$. For the inductive step, we assume that q^{s-1} is a quasi-isomorphism for some $s \geq 1$. We have a commutative diagram of complexes

$$\begin{array}{ccccccccc} 0 & \longrightarrow & F^{s-1} & \longrightarrow & F^s & \longrightarrow & \mathrm{gr}^s(F) & \longrightarrow & 0 \\ & & \downarrow q^{s-1} & & \downarrow q^s & & \downarrow \mathrm{gr}^s(q) & & \\ 0 & \longrightarrow & F'^{s-1} & \longrightarrow & F'^s & \longrightarrow & \mathrm{gr}^s(F') & \longrightarrow & 0 \end{array}$$

By hypothesis and inductive assumption, in the induced diagram

$$\begin{array}{ccccccccc} H_{n+1}(\mathrm{gr}^s(F)) & \rightarrow & H_n(F^{s-1}) & \rightarrow & H_n(F^s) & \rightarrow & H_n(\mathrm{gr}^s(F)) & \rightarrow & H_{n-1}(F^{s-1}) \\ H_{n+1}(\mathrm{gr}^s(q)) \downarrow \cong & & H_n(q^{s-1}) \downarrow \cong & & H_n(q^s) \downarrow & & H_n(\mathrm{gr}^s(q)) \downarrow \cong & & H_{n-1}(q^{s-1}) \downarrow \cong \\ H_{n+1}(\mathrm{gr}^s(F')) & \rightarrow & H_n(F'^{s-1}) & \rightarrow & H_n(F'^s) & \rightarrow & H_n(\mathrm{gr}^s(F')) & \rightarrow & H_{n-1}(F'^{s-1}) \end{array}$$

the four outer vertical maps are bijective. By the Five-Lemma, so is $H_n(q^s)$.

Now we fix an integer $n \in \mathbb{Z}$, and pick s so large that

$$F_k^s = F_k \quad \text{and} \quad F_k'^s = F_k' \quad \text{hold for} \quad k = n-1, n, n+1.$$

The choice of s implies that $H_n(F^s) = H_n(F)$, $H_n(F'^s) = H_n(F')$, and $H_n(q^s) = H_n(q)$. Since we have already proved that $H_n(q^s)$ is an isomorphism, we conclude that $H_n(q): H_n(F) \rightarrow H_n(F')$ is an isomorphism. \square

Remark 3.3. If (F, d_F) is a complex of A -modules, then $\text{Hom}_A^{\text{gr}}(F, F)$ denotes the graded module whose n 'th component consists of the A -linear maps $g: F \rightarrow F$ with $g(F_i) \subseteq F_{i+n}$ for all $i \in \mathbb{Z}$.

If g, h are homogeneous A -linear maps, then their composition gh is homogeneous of degree $\deg(g) + \deg(h)$, and so is their graded commutator

$$[g, h] = gh - (-1)^{\deg g \cdot \deg h} hg$$

Commutation is a graded derivation: for each homogeneous map h' one has

$$[g, hh'] = [g, h]h' + (-1)^{\deg g \cdot \deg h} h[g, h']$$

The map $h \mapsto [d_F, h]$ has square 0, and transforms $\text{Hom}_A^{\text{gr}}(F, F)$ into a complex of A -modules; by definition, its cycles are the chain maps $F \rightarrow F$, and its boundaries are the null-homotopic maps.

Remark 3.4. If $p: F \rightarrow F'$ and $q: G \rightarrow G'$ are graded maps of graded modules, we define the tensor product $p \otimes q: F \otimes F' \rightarrow G \otimes G'$ by the formula $(p \otimes q)(f \otimes g) = (-1)^{\deg q \cdot \deg f} (p(f) \otimes q(g))$. With this convention, when $F = F'$ and $G = G'$, the graded commutator $[1_F \otimes q, p \otimes 1_G]$ vanishes.

Lemma 3.5. *Let M be an A -module and let $r: C \rightarrow M$ be a free resolution. If $g: C \rightarrow C$ is an A -linear map with $\deg(g) > 0$, and $[d_C, g] = 0$, then $g = [d_C, h]$ for some A -linear map $h: C \rightarrow C$ with $\deg(h) = \deg(g) + 1$.*

Proof. The augmentation $r: C \rightarrow M$ defines a chain map of degree zero

$$\text{Hom}_A^{\text{gr}}(C, r): \text{Hom}_A^{\text{gr}}(C, C) \rightarrow \text{Hom}_A^{\text{gr}}(C, M)$$

The map induced in homology is an isomorphism: to see this, apply the ‘comparison theorem for projective resolutions’. Since A -linear maps $C \rightarrow M$ of positive degree are trivial, the conclusion follows from Remark 3.3. \square

Proof (of Theorem 3.1). Recall that D is the divided powers algebra of a free A -module Q with basis Y_1, \dots, Y_c , that X_1, \dots, X_c is the dual basis of the free A -module Q^* , and S for the symmetric algebra of Q^* , see Construction 2.1 for details. We set $f = \sum_{i=1}^c f_i X_i \in S^1(Q^*)$.

We first construct the maps d_γ by induction on $|\gamma|$.

If $|\gamma| = 0$, then $\gamma = o$, so $d_o = d_C$ is predefined. If $|\gamma| = 1$, then $\gamma = \epsilon_i$ for some i with $1 \leq i \leq c$. Since f_i annihilates the B -module M , the map $-f_i \cdot 1_C$ lifts the zero map on M , hence is null-homotopic. For each i we take d_{ϵ_i} to be a null-homotopy, that is, $[d_o, d_{\epsilon_i}] = -f_i \cdot 1_C$. With these choices, the desired formulas hold for all γ with $|\gamma| \leq 1$.

Assume by induction that maps d_γ satisfying the conclusion of the lemma have been chosen for all $\gamma \in \mathbb{N}^c$ with $|\gamma| < n$, for some $n \geq 2$. Fix $\gamma \in \mathbb{N}^c$

with $|\gamma| = n$. Using Remark 3.3 and the induction hypothesis, we obtain

$$\begin{aligned}
\left[d_o, \sum_{\alpha+\beta=\gamma}^+ d_\alpha d_\beta \right] &= \sum_{\alpha+\beta=\gamma}^+ ([d_o, d_\alpha] d_\beta - d_\alpha [d_o, d_\beta]) \\
&= \sum_{\alpha+\beta=\gamma}^+ \left(\left(\sum_{\alpha'+\alpha''=\alpha}^+ d_{\alpha'} d_{\alpha''} \right) d_\beta - d_\alpha \left(\sum_{\beta'+\beta''=\beta}^+ d_{\beta'} d_{\beta''} \right) \right) \\
&= \sum_{\alpha'+\alpha''+\beta=\gamma}^+ d_{\alpha'} d_{\alpha''} d_\beta - \sum_{\alpha+\beta'+\beta''=\gamma}^+ d_\alpha d_{\beta'} d_{\beta''} \\
&= 0
\end{aligned}$$

The map $-\sum_{\alpha+\beta=\gamma}^+ d_\alpha d_\beta$ has degree $2|\gamma| - 2$, so by Lemma 3.5 it is equal to $[d_o, d_\gamma]$ for some A -linear map $d_\gamma: C \rightarrow C$ of degree $2|\gamma| - 1$. Choosing such a d_γ for each $\gamma \in \mathbb{N}^c$ with $|\gamma| = n$, we complete the step of the induction.

From the definition of d we obtain an expression

$$\begin{aligned}
d^2 &= d_{CD}^2 \otimes 1_E + 1_C \otimes [d_{DE}, 1_D \otimes d_E] + [d_{CD} \otimes 1_E, 1_C \otimes d_{DE}] + \\
&\quad 1_C \otimes d_{DE}^2 + 1_C \otimes 1_D \otimes d_E^2 + [d_{CD} \otimes 1_E, 1_C \otimes 1_D \otimes d_E]
\end{aligned}$$

Constructions 2.4, 2.2, and Remark 3.4 show that the maps in the second row are equal to 0, so to prove that $d^2 = 0$ it suffices to establish the equalities

$$d_{CD}^2 = -f \cdot 1_{C \otimes D} \quad (4)$$

$$[d_{DE}, 1_D \otimes d_E] = f \cdot 1_{D \otimes E} \quad (5)$$

$$[d_{CD} \otimes 1_E, 1_C \otimes d_{DE}] = 0 \quad (6)$$

A direct computation with formula (1) proves equality (4) above:

$$\begin{aligned}
d_{CD}^2(x \otimes y) &= d_{CD} \left(\sum_{\beta \in \mathbb{N}^c} d_\beta(x) \otimes (X^\beta \lrcorner y) \right) \\
&= \sum_{\beta \in \mathbb{N}^c} \left(\sum_{\alpha \in \mathbb{N}^c} d_\alpha d_\beta(x) \otimes (X^\alpha \lrcorner (X^\beta \lrcorner y)) \right) \\
&= \sum_{\alpha+\beta \in \mathbb{N}^c} d_\alpha d_\beta(x) \otimes (X^{\alpha+\beta} \lrcorner y) \\
&= \sum_{i=1}^c -f_i x \otimes (X_i \lrcorner y) \\
&= -f \cdot (x \otimes y)
\end{aligned}$$

By Constructions 2.2 and 2.4, the maps $1_D \otimes d_E$ and d_{DE} are derivations of degree -1 , so the commutator $[d_{DE}, 1_D \otimes d_E]$ is a derivation of degree -2 . Every element of $D \otimes_A E$ is a product of elements $1 \otimes Y_i^\wedge$ of degree 1 and $Y_j^{(k)} \otimes 1$ of degree $2k$, so it suffices to check that the map on either side of

(6) takes the same value on those elements. For degree reasons, both sides vanish on $1 \otimes Y_i^\wedge$. We now complete the proof of equality (5) as follows:

$$\begin{aligned}
& [d_{DE}, 1_E \otimes d_E](Y_j^{(k)} \otimes 1) \\
&= d_{DE}((1_D \otimes d_E)(Y_j^{(k)} \otimes 1)) + (1_E \otimes d_E)(d_{DE}(Y_j^{(k)} \otimes 1)) \\
&= (1_E \otimes d_E)(Y_j^{(k-1)} \otimes Y_j^\wedge) \\
&= Y_j^{(k-1)} \otimes f_j \\
&= f \cdot (Y_j^{(k)} \otimes 1)
\end{aligned}$$

To derive equation (6) we use Constructions 2.2 and 2.4 once again:

$$\begin{aligned}
& ((d_{CD} \otimes 1_E)(1_C \otimes d_{DE}))(x \otimes y \otimes z) \\
&= (-1)^{\deg x} d_{CD} \left(\sum_{i=1}^c x \otimes (X_i \lrcorner y) \otimes (Y_i^\wedge \cdot z) \right) \\
&= (-1)^{\deg x} \sum_{i=1}^c \sum_{\gamma \in \mathbb{N}^c} d_\gamma(x) \otimes (X^\gamma \lrcorner (X_i \lrcorner y)) \otimes (Y_i^\wedge \cdot z) \\
&= - \sum_{\gamma \in \mathbb{N}^c} \sum_{i=1}^c (-1)^{\deg(d_\gamma(x))} d_\gamma(x) \otimes (X_i \lrcorner (X^\gamma \lrcorner y)) \otimes (Y_i^\wedge \cdot z) \\
&= - (1_C \otimes d_{DE}) \left(\sum_{\gamma \in \mathbb{N}^c} d_\gamma(x) \otimes (X^\gamma \lrcorner y) \otimes z \right) \\
&= - ((1_C \otimes d_{DE})(d_{CD} \otimes 1_E))(x \otimes y \otimes z)
\end{aligned}$$

It remains to show q is a quasi-isomorphism. Setting

$$F^s = \bigoplus_{k+\ell \leq s} C \otimes_A D^k(Q) \otimes_A \bigwedge^\ell(Q^\wedge) \quad \text{for } s \in \mathbb{Z}$$

we obtain a bounded filtration of the complex $F = (C \otimes_A D \otimes_A E, d)$. On the other hand, we let F' denote the complex with $F'_0 = M$ and $F'_n = 0$ for $n \neq 0$; the filtration defined by $F'^0 = 0$ and $F'^s = F'$ for $s \geq 1$ is obviously bounded, and $q(F^s) \subseteq F'^s$ holds for all $s \geq 0$. By Lemma 3.2 it suffices to show that the induced map $\text{gr}^s(q): \text{gr}^s(F) \rightarrow \text{gr}^s(F')$ is bijective for all s .

Inspection of the differential d of F shows that $\text{gr}^s(F)$ is isomorphic to the tensor product of complexes $C \otimes_A G^s$, where G^s is the complex defined in Lemma 2.5. It is established there that G^s is split exact for $s > 0$, hence $H_n(C \otimes_A G^s) = 0$ for all $n \in \mathbb{Z}$. As $G^0 = A$ and $\text{gr}^0(q) = r$, we are done. \square

4 Cohomology Operators

We present a new approach to the procedure of Shamash [15] and Eisenbud [11] for building projective resolutions over a complete intersection. We then

use this resolution to prove a fundamental result of Gulliksen [12] on the structure of Ext modules over complete intersections.

A set $\mathbf{f} = \{f_1, \dots, f_c\} \subseteq A$ is *Koszul-regular* if the complex (E, d_E) of Construction 2.2, has $H_n(E) = 0$ for $n > 0$. A sufficient condition for Koszul-regularity is that the elements of \mathbf{f} , in some order, form a regular sequence.

Theorem 4.1. *Let A be a commutative ring, $\mathbf{f} = \{f_1, \dots, f_c\} \subseteq A$ a subset, $B = A/(\mathbf{f})$ the residue ring, M a B -module, and $r: C \rightarrow M$ a resolution of M by projective (respectively, free) A -modules.*

Let $\{d_\gamma: C \rightarrow C\}_{\gamma \in \mathbb{N}^c}$ be a family of A -linear maps provided by Theorem 3.1, set $D' = D \otimes_A B$, and $y' = y \otimes 1$ for $y \in D$. The map

$$\begin{aligned} \partial: C \otimes_A D' &\rightarrow C \otimes_A D' && \text{given by} \\ \partial(x \otimes y') &= \sum_{\gamma \in \mathbb{N}^c} d_\gamma(x) \otimes (X^\gamma \lrcorner y') \end{aligned} \tag{7}$$

is a B -linear differential of degree -1 . If \mathbf{f} is Koszul-regular, then the map

$$\begin{aligned} q': C \otimes_A D' &\rightarrow M && \text{given by} \\ q'(x \otimes y') &= \begin{cases} y \cdot r(x) & \text{if } \deg(y') = 0 \\ 0 & \text{otherwise} \end{cases} \end{aligned}$$

is a resolution of M by projective (respectively, free) A -modules.

Remark 4.2. Assume that in the theorem $\mathbf{f} = \{f_1\}$. The module D_ℓ is then trivial if ℓ is odd, and is free with basis consisting of a single element $Y_1^{(\ell/2)}$ if ℓ is even. Thus, the resolution $C \otimes_A D'$ has the form

$$\dots \xrightarrow{\partial_{2n+1}} \bigoplus_{j=0}^{\infty} C_{2j} \otimes_A B Y_1^{(n-j)} \xrightarrow{\partial_{2n}} \bigoplus_{j=1}^{\infty} C_{2j-1} \otimes_A B Y_1^{(n-j)} \xrightarrow{\partial_{2n-1}} \dots$$

The simplest situation occurs when, in addition, C is a free resolution with $C_n = 0$ for $n \geq 2$. In this case the differential d_o has a single non-zero component, $d_1: C_1 \rightarrow C_0$, the homotopy d_{ϵ_1} between $-f \cdot 1_C$ and 0_C has a single non-zero component, $s_0: C_0 \rightarrow C_1$, and all the maps d_γ with $\gamma \in \mathbb{N}^1 \setminus \{o, \epsilon_1\}$ are trivial for degree reasons. It is now easy to see that the complex above coincides with the one constructed, *ad hoc*, in Remark 1.6.

Proof (of the theorem). In the notation of Theorem 3.1, we have equalities

$$C \otimes_A D' = (C \otimes_A D \otimes_A E) \otimes_E B \quad \text{and} \quad \partial = d \otimes 1_B$$

It follows that $\partial^2 = 0$. For each $s \geq 0$ consider the subcomplexes

$$\begin{aligned} F^s &= \bigoplus_{k+\ell \leq s} C_k \otimes_A D_\ell \otimes_A E && \text{of} && F = C \otimes_A D \otimes_A E \\ F'^s &= \bigoplus_{k+\ell \leq s} C_k \otimes_A D'_\ell && \text{of} && F' = C \otimes_A D' \end{aligned}$$

They provide bounded filtrations of the complexes F and F' , respectively, such that the map $p' = 1_C \otimes 1_D \otimes p: F \rightarrow F'$ satisfies $p'(F^s) \subseteq F'^s$ for all $s \geq 0$. Setting $G_s = \bigoplus_{k+\ell=s} (C_k \otimes_A D_\ell)$, we obtain equalities $\text{gr}^s(F) = (G_s \otimes_A E, 1_{G_s} \otimes d_E)$ and $\text{gr}^s(F') = (G_s \otimes_A B, 0)$ of complexes of A -modules.

If \mathbf{f} is Koszul regular, then $p: E \rightarrow B$ is a quasi-isomorphism, hence so is $1_{G_s} \otimes p = \text{gr}^s(p')$ for each $s \geq 0$. Lemma 3.2 then shows that p' is a quasi-isomorphism. The quasi-isomorphism $q: F \rightarrow M$ of Theorem 3.1 factors as $q = q'(1_C \otimes 1_D \otimes p)$, so we see that q' is a quasi-isomorphism, as desired. \square

Let M and N be B -modules, and let $\text{Ext}_B^\bullet(M, N)$ denote the graded B -module having $\text{Ext}_B^n(M, N)$ as component of degree $-n$. To avoid negative numbers, it is customary to regrade $\text{Ext}_B^\bullet(M, N)$ by cohomological degree, under which the elements of $\text{Ext}_B^n(M, N)$ are assigned degree n ; we do not do it here, in order not to confuse *Macaulay 2*. Of course, these modules can be computed from any projective resolution of M over B .

The next couple of remarks collect a few innocuous observations. In hindsight, they provide some of the basic tools for studying cohomology of modules over complete intersections: see Remark 4.6 for some related material.

Remark 4.3. The resolution $(C \otimes_A D', \partial)$ provided by Theorem 4.1 is a graded module over the graded algebra S , with action defined by the formula

$$u \cdot (x \otimes y') = x \otimes (u \lrcorner y)'$$

and this action commutes with the differential ∂ . The induced action provides a structure of graded S -module on the complex $\text{Hom}_B(C \otimes_A D', N)$.

The action of S commutes with the differential $\partial^* = \text{Hom}_B(\partial, N)$ of this complex, hence passes to its homology, making it a graded S -module. Thus, each element $u \in S_{-2k} = S^k(Q)$ determines homomorphisms

$$\text{Ext}_B^n(M, N) \xrightarrow{u} \text{Ext}_B^{n+2k}(M, N) \quad \text{for all } n \in \mathbb{Z}$$

For this reason, from now on we refer to the graded ring S as the *ring of cohomology operators* determined by the Koszul-regular set \mathbf{f} .

Remark 4.4. The canonical isomorphisms of complexes of A -modules

$$\text{Hom}_B(C \otimes_A D', N) = S \otimes_A \text{Hom}_A(C, N) = S \otimes_A \text{Hom}_A(C, A) \otimes_A N$$

commute with the actions of S .

The following fundamental result shows that in many important cases the action of the cohomology operators is highly nontrivial.

Theorem 4.5. *Let A be a commutative ring, let \mathbf{f} be a Koszul regular subset of A , and let S be the graded ring of cohomology operators defined by \mathbf{f} .*

If M and N are finitely generated modules over $B = A/(\mathbf{f})$, and M has finite projective dimension over A (in particular, if A is regular), then the graded S -module $\text{Ext}_B^\bullet(M, N)$ is finitely generated.

Proof. Choose a resolution $r: C \rightarrow M$ with C_n a finite projective A -module for each n and $C_n = 0$ for all $n \gg 0$. By Remark (4.4), the graded S -module $\mathrm{Hom}_B(C \otimes_A D', N)$ is finitely generated. Since S is noetherian, so is the submodule $\mathrm{Ker}(\partial^*)$, and hence the homology module, $\mathrm{Ext}_B^\bullet(M, N)$. \square

Remark 4.6. The resolution of Remark 4.2 was constructed by Shamash [15, Sect. 3], that of Theorem 4.1 by Eisenbud [11, Sect. 7]. The new aspect of our approach is indicated at the beginning of Section 3.

As introduced in Remark 4.3, the S -module structure on Ext may appear *ad hoc*. In fact, it is independent of all choices of resolutions and maps, it can be computed from *any* projective resolution of M over B , it is natural in both module arguments and—in an appropriate sense—in the ring argument, and it commutes with Yoneda products from either side. These properties were proved by Gulliksen [12, Sect. 2], Mehta [13, Ch. 2], Eisenbud [11, Sect. 4], and Avramov [2, Sect. 2]. However, each author used a different construction of cohomology operators, and comparison of the different approaches has turned to be an unexpectedly delicate problem. It was finally resolved in [8], where complete proofs of the main properties of the operators can be found.

Gulliksen [12, Sect. 3] established a stronger form of Theorem 4.5, without finiteness hypotheses on the ring A : If the A -module $\mathrm{Ext}_A^\bullet(M, N)$ is noetherian, then the S -module $\mathrm{Ext}_B^\bullet(M, N)$ is noetherian; this can be obtained from the complexes of Remark (4.4) by means of a spectral sequence, cf. [4, Sect. 6]. The converse of Gulliksen's theorem was proved in [6, Sect. 4].

For the rest of the paper we place ourselves in a situation where *Macaulay 2* operates best—graded modules over positively graded rings. This grading is inherited by the various Ext modules, and we keep careful track of it. Our conventions and bookkeeping procedures are discussed in detail in an Appendix, which the reader is invited to consult as needed.

For ease of reference, we collect some notation.

Notation 4.7. The following is assumed for the rest of the paper.

- K is a field.
- $\{x_h \mid \deg'(x_h) > 0\}_{h=1, \dots, e}$ is a set of indeterminates over K .
- $A = K[x_1, \dots, x_e]$, graded by $\deg'(a) = 0$ for $a \in K$.
- f_1, \dots, f_c is a homogeneous A -regular sequence in $(x_1, \dots, x_e)^2$.
- $r_i = \deg'(f_i)$ for $i = 1, \dots, c$.
- $\{X_i \mid \mathrm{Deg} X_i = (-2, -r_i)\}_{i=1, \dots, c}$ is a set of indeterminates over A .
- $S = A[X_1, \dots, X_c]$, bigraded by $\mathrm{Deg}(a) = (0, \deg'(a))$.
- $B = A/(\mathbf{f})$, with degree induced by \deg' .
- M and N are finitely generated graded B -modules.
- S acts as bigraded ring of cohomology operators on $\mathrm{Ext}_B^\bullet(M, N)$.
- $k = B/(x_1, \dots, x_e)B$, with degree induced by \deg' .
- $R = S \otimes_A k \cong K[X_1, \dots, X_c]$, with bidegree induced by Deg .

Remark 4.8. Under the conditions above, it is reasonable to ask when the B -free resolution G of Theorem 4.1, obtained from a *minimal* A -free resolution C of M , will itself be minimal. Shamash [15, Sect. 3] proves that G is minimal if $f_i \in (x_1, \dots, x_e)\text{ann}_A(M)$ for $i = 1, \dots, c$. An obvious example with non-minimal G occurs when M has finite projective dimension over B : if $c > 0$ then G is infinite. A more interesting failure of minimality follows.

Example 4.9. Let A, f, B , and M be as in Example 1.5.

```
i38 : M = B^1/m^3;
i39 : F = resolution(M, LengthLimit=>8)
      1      10      16      15      15      15      15      15      15
o39 = B <-- B <-- B <-- B <-- B <-- B <-- B <-- B <-- B
      0      1      2      3      4      5      6      7      8
o39 : ChainComplex
```

Thus, the sequence of Betti numbers $\beta_n^B(M)$ is $(1, 10, 16, 15, 15, 15, \dots)$.

```
i40 : M' = restrict1 M;
i41 : C = res M'
      1      10      15      6
o41 = A <-- A <-- A <-- A <-- 0
      0      1      2      3      4
o41 : ChainComplex
```

By Remark 4.2, the sequence $\text{rank}_B F_n$ is $(1, 10, 16, 16, 16, 16, \dots)$.

In a graded context, all cohomological entities discussed so far in the text acquire an extra grading, discussed in detail in the Appendix. The notions below are used, but not named, in [5] in a local situation.

Remark 4.10. We define the *reduced Ext module* for M and N over B by

$$\text{ext}_B^\bullet(M, N) = \text{Ext}_B^\bullet(M, N) \otimes_A k$$

With the induced bigrading and action, it is a bigraded module over the bigraded ring R , that we call the *reduced ring of cohomology operators*.

The dimension of the K -vector space $\text{ext}_B^n(M, N)_s$ is equal to the number of generators of bidegree $(-n, s)$ in any minimal set of generators of the graded B -module $\text{Ext}_B^n(M, N)$. We define the graded (respectively, ungraded) *Ext-generator series* of M and N to be the formal power series

$$G_B^{M,N}(t, u) = \sum_{n \in \mathbb{N}, s \in \mathbb{Z}} \dim_K \text{ext}_B^n(M, N)_s t^n u^{-s} \in \mathbb{Z}[u, u^{-1}][[t]]$$

$$G_B^{M,N}(t) = \sum_{n=0}^{\infty} \dim_K \text{ext}_B^n(M, N) t^n \in \mathbb{Z}[[t]]$$

There is a simple relation between these series: $G_B^{M,N}(t) = G_B^{M,N}(t, 1)$.

Corollary 4.11. *In the notation above, $\text{ext}_B^\bullet(M, N)$ is a finitely generated bigraded R -module, and $G_B^{M, N}(t, u)$ represents a rational function of the form*

$$\frac{g_B^{M, N}(t, u)}{(1 - t^2 u^{r_1}) \cdots (1 - t^2 u^{r_c})} \quad \text{with} \quad g_B^{M, N}(t, u) \in \mathbb{Z}[t, u, u^{-1}]$$

Proof. The assertion on finite generation results from Theorem 4.5 and the one on bigradings from Remark A.2. The form of the power series is then given by the Hilbert-Serre Theorem. \square

5 Computation of Ext Modules

This section contains the main new computational result of the paper.

We discuss, apply, and present an algorithm that computes, for graded modules M and N over a graded complete intersection ring B , the graded B -modules $\text{Ext}_B^n(M, N)$ simultaneously in all degrees n , along with all the cohomology operators defined in Remark 4.3.

More precisely, the input consists of a field K , a polynomial ring $A = K[x_1, \dots, x_e]$ with $\deg'(x_h) > 0$, a sequence f_1, \dots, f_c of elements of A , and finitely generated modules M, N over $B = A/(f_1, \dots, f_c)$. The program checks whether the sequence consists of homogeneous elements, whether it is regular, and whether the modules M, N are graded, sending the appropriate error message if any one of these conditions is violated. If the input data pass those tests, then the program produces a presentation of the bigraded module $H = \text{Ext}_B^\bullet(M, N)$, where the elements of $\text{Ext}_B^n(M, N)$ have homological degree $-n$, over the polynomial ring $A[X_1, \dots, X_c]$, bigraded by $\text{Deg}(a) = (0, \deg'(a))$ and $\text{Deg}(X_i) = (-2, -\deg'(f_i))$.

The algorithm is based on the proofs of Theorems 4.1 and 4.5, and is presented in Code 5.4 below. We start with an informal discussion.

Remark 5.1. The routine `resolution` of *Macaulay 2* finds the matrices $d_{o, n}: C_n \rightarrow C_{n-1}$ of the differential d_C of a minimal free resolution C of M over A . Matrices $d_{\gamma, n}: C_n \rightarrow C_{n+2|\gamma|-1}$ satisfying equation (1) for $\gamma \in \mathbb{N}^c$ with $|\gamma| > 0$ are computed using the routine `nullhomotopy` of the *Macaulay 2* language in a loop that follows the first part of the proof of Theorem 4.1.

The transposed matrix $d_{\gamma, n}^*$ yields an endomorphism of the free bigraded A -module $C^* = \bigoplus_{n=0}^e \text{Hom}_A(C_n, A)$ of rank m , where $m = \sum_{n=0}^e \text{rank}_A C_n$. The $m \times m$ matrix $\tilde{d}_{\gamma, n}^*$ describing this endomorphism is formed using the routines `transpose` and `sum`. The $m \times m$ matrix

$$\Delta = \sum_{n=0}^e (-1)^{n+2|\gamma|-1} \sum_{\substack{\gamma \in \mathbb{N}^c \\ |\gamma| \leq (e-n+1)/2}} X^\gamma \cdot \tilde{d}_{\gamma, n}^*$$

with entries in $S = A[X_1, \dots, X_c]$ defines an endomorphism of the free bigraded S -module $S \otimes_A C^*$. It induces an endomorphism $\overline{\Delta}$ of the bigraded

S -module $S \otimes_A C^* \otimes_A N$. The bigraded S -module $H = \text{Ext}_B^\bullet(M, N)$ is computed as $H = \text{Ker}(\overline{\Delta})/\text{Im}(\overline{\Delta})$ using the routine `homology`.

In the computations we let H denote the bigraded S -module $\text{Ext}_B^\bullet(M, N)$. As the graded ring S is zero in odd homological degrees, there is a canonical direct sum decomposition $H = H^{\text{even}} \oplus H^{\text{odd}}$ of bigraded S -modules, where ‘even’ or ‘odd’ refers to the parity of the *first* degree in each pair $\text{Deg}(x)$.

We begin with an example in codimension 1, where it is possible to construct the infinite resolution and the action of S on it by hand.

Example 5.2. Consider the ring $A = K[x]$ where the variable x is assigned degree 5, and set $B = A/(x^3)$. The bigraded ring of cohomology operators then is $S = A[X, x]$, where $\text{Deg}(X) = (-2, -15)$ and $\text{Deg } x = (0, 5)$.

For the B -modules $M = B/(x^2)$ and $N = B/(x)$, the bigraded S -module $H = \text{Ext}_B^\bullet(M, N)$ is described by the isomorphism

$$H \cong (S/(x)) \oplus (S/(x))[1, 10]$$

A minimal free resolution of M over B is displayed below.

$$F = \dots \longrightarrow B[-30] \xrightarrow{-x} B[-25] \xrightarrow{x^2} B[-15] \xrightarrow{-x} B[-10] \xrightarrow{x^2} B \longrightarrow 0$$

This resolution is actually isomorphic to the resolution $C \otimes_A D'$ described in Remark 4.2, formed from the free resolution

$$C = 0 \longrightarrow A[-10] \xrightarrow{x^2} A \longrightarrow 0$$

of M over A and the nullhomotopy d_{ϵ_1} displayed in the diagram

$$\begin{array}{ccccccc} 0 & \longrightarrow & A[-10] & \xrightarrow{x^2} & A & \longrightarrow & 0 \\ & & \downarrow -x^3 & \swarrow -x & \downarrow -x^3 & & \\ 0 & \longrightarrow & A[-10] & \xrightarrow{x^2} & A & \longrightarrow & 0 \end{array}$$

The isomorphism of F with $C \otimes_A D'$ endows F with a structure of bigraded module over S , where the action of X on F is the chain map $F \rightarrow F$ of homological degree -2 and internal degree -15 that corresponds to the identity map of B in each component.

The bigraded S -module $H = \text{Ext}_B^\bullet(M, N)$ is the homology of the complex

$$\text{Hom}_R(F, N) = 0 \longrightarrow N \xrightarrow{0} N[10] \xrightarrow{0} N[15] \xrightarrow{0} N[25] \xrightarrow{0} N[30] \longrightarrow \dots$$

where multiplication by $x \in S$ is the zero map, and for each $i \geq 0$ multiplication by $X \in S$ sends $N[15i]$ to $N[15i + 15]$ (respectively, $N[10i + 10]$ to $N[10i + 25]$) by the identity map. This description provides the desired isomorphisms of bigraded S -modules.

Here is how to compute H with *Macaulay 2*.

Create the rings and modules.

```

i42 : K = ZZ/103;
i43 : A = K[x,Degrees=>{5}];
i44 : B = A/(x^3);
i45 : M = B^1/(x^2);
i46 : N = B^1/(x);

```

Use the function `Ext` to compute $H = \text{Ext}_B^*(M, N)$ (the semicolon at the end of the line will suppress printing until we have assigned the name `S` to the ring of cohomology operators constructed by *Macaulay 2*.)

```

i47 : H = Ext(M,N);

```

We may look at the ring.

```

i48 : ring H
o48 = K [$X , x, Degrees => {{-2, -15}, {0, 5}}]
      1
o48 : PolynomialRing

```

Macaulay 2 has assigned the name `$X_1` to the variable X . The dollar sign indicates an internal name that cannot be entered from the keyboard; if necessary, obtain the variable by entering `S_0`; notice that indexing in *Macaulay 2* starts with 0 rather than 1. Notice also the appearance of braces rather than parentheses in *Macaulay 2*'s notation for bidegrees.

```

i49 : degree \ gens ring H
o49 = {{-2, -15}, {0, 5}}
o49 : List

```

Assign the ring a name.

```

i50 : S = ring H;

```

We can now look at the S -module H .

```

i51 : H
o51 = cokernel {0, 0} | 0 x |
              {-1, -10} | x 0 |
                        2
o51 : S-module, quotient of S

```

Each row in the display above is labeled with the bidegree of the corresponding generator of H . This presentation gives the isomorphisms of bigraded S -modules, already computed by hand earlier.

Let's try an example with a complete intersection of codimension 2. It is not so easy to do by hand, but can be checked using the theory in [4].

Example 5.3. Begin by constructing a polynomial ring $A = K[x, y]$.

```

i52 : A = K[x,y];

```

Now we produce a complete intersection quotient ring $B = A/(x^3, y^2)$.

```

i53 : J = ideal(x^3,y^2);
o53 : Ideal of A
i54 : B = A/J;
We take N to be the B-module B/(x^2,xy).
i55 : N = cokernel matrix{{x^2,x*y}}
o55 = cokernel | x2 xy |

```

Remark A.1 shows that $H = \text{Ext}_B^\bullet(N, N)$ is a bigraded module over the bigraded ring $S = A[X_1, X_2] = K[X_1, X_2, x, y]$ where

$$\begin{aligned} \text{Deg}(X_1) &= (-2, -3) & \text{Deg}(X_2) &= (-2, -2) \\ \text{Deg}(x) &= (0, 1) & \text{Deg}(y) &= (0, 1) \end{aligned}$$

Using *Macaulay 2* (below) we obtain an isomorphism of bigraded S -modules

$$\begin{aligned} H^{\text{even}} &\cong \frac{S}{(x^2, xy, y^2, xX_1, yX_1)} \oplus \frac{S}{(x, y)}[2, 2] \\ H^{\text{odd}} &\cong \left(\frac{S}{(x, y, X_1)} \oplus \frac{S}{(x, y)} \right)^2 [1, 1] \end{aligned}$$

These isomorphisms also yield expressions for the graded B -modules:

$$\begin{aligned} \text{Ext}_B^{2i}(N, N) &\cong N \cdot X_2^i \oplus \bigoplus_{h=1}^i k \cdot X_1^h X_2^{i-h} \oplus \bigoplus_{h=0}^{i-1} k[2] \cdot X_1^h X_2^{i-1-h} \\ \text{Ext}_B^{2i+1}(N, N) &\cong \left(k[1] \cdot X_2^i \oplus \bigoplus_{h=0}^i k[1] \cdot X_1^{i-h} X_2^h \right)^2 \end{aligned}$$

Now we follow in detail the computation of the bigraded S -module H .

```

i56 : time H = Ext(N,N);
      -- used 0.2 seconds
i57 : ring H
o57 = K [$X , $X , x, y, Degrees => {{-2, -2}, {-2, -3}, {0, 1}, {0, 1}}]
      1      2
o57 : PolynomialRing
i58 : S = ring H;

```

One might wish to have a better view of the bidegrees of the variables of the ring S . An easy way to achieve this, with signs reversed, is to display the transpose of the matrix of variables.

```

i59 : transpose vars S

```

```

o59 = {2, 2} | $X_1 |
      {2, 3} | $X_2 |
      {0, -1} | x   |
      {0, -1} | y   |

          4      1
o59 : Matrix S <--- S

```

The internal degrees displayed for the cohomology operators may come as a surprise. To understand what is going on, recall that these degrees are determined by a choice of minimal generators for J . At this point we do not know what is the sequence of generators that *Macaulay 2* used, so let's compute those generators the way the program did.

```

i60 : trim J

          2  3
o60 = ideal (y , x )

o60 : Ideal of A

```

Notice that *Macaulay 2* has reordered the original sequence of generators. Now we see that our variable X_1 , which corresponds to x^3 , is denoted X_2 by *Macaulay 2*, and that X_2 , which corresponds to y^2 is denoted X_1 . This explains the bidegrees used by the program.

Display H .

```

i61 : H

o61 = cokernel {-2, -2} | 0 0 0 0 0 0 0 0 0 0 0 0 y x 0 0 0 ...
                  {-1, -1} | y 0 0 0 0 x 0 0 0 0 0 0 0 0 $X_1 0 0 ...
                  {-1, -1} | 0 0 0 y 0 0 0 x 0 0 0 0 0 0 0 0 $X_1 0 ...
                  {-1, -1} | 0 y 0 0 x 0 0 0 0 0 0 0 0 0 0 0 0 ...
                  {-1, -1} | 0 0 y 0 0 0 x 0 0 0 0 0 0 0 0 0 0 ...
                  {0, 0}   | 0 0 0 0 0 0 0 0 0 y2 xy x2 0 0 0 0 0 $X_1y ...

          6
o61 : S-module, quotient of S

```

That's a bit large, so we want to look at the even and odd parts separately.

We may compute the even and odd parts of H as the span of the generators of H with the appropriate parity. Since the two desired functions differ only in the predicate to be applied, we can generate them both by writing a function that accepts the predicate as its argument and returns a function.

```

i62 : partSelector = predicate -> H -> (
      R := ring H;
      H' := prune image matrix {
        select(
          apply(numgens H, i -> H_{i}),
          f -> predicate first first degrees source f
        )
      };
      H');

i63 : evenPart = partSelector even; oddPart = partSelector odd;

Now to obtain the even part,  $H^{\text{even}}$ , simply type

i65 : evenPart H

```



```
o65 = cokernel {-2, -2} | 0 0 0 y x 0 0 |
              {0, 0} | y2 xy x2 0 0 $X_1y $X_1x |
```

2

```
o65 : S-module, quotient of S
```

Do the same thing for the odd part, H^{odd} .

```
i66 : oddPart H
```

```
o66 = cokernel {-1, -1} | 0 0 y 0 0 0 x 0 0 0 |
              {-1, -1} | 0 y 0 0 x 0 0 0 0 |
              {-1, -1} | 0 0 0 y 0 0 0 x 0 $X_1 |
              {-1, -1} | y 0 0 0 0 x 0 0 $X_1 0 |
```

4

```
o66 : S-module, quotient of S
```

These presentations yield the desired isomorphism of bigraded S -modules.

Here is the source code which implements the routine `Ext`. It is incorporated into *Macaulay 2*.

Code 5.4. The function `Ext(M,N)` computes $\text{Ext}_B^\bullet(M, N)$ for graded modules M, N over a graded complete intersection ring B . The function code can be used to obtain a copy of the source code.

```
i67 : print code(Ext,Module,Module)
-- ../../m2/ext.m2:82-171
Ext(Module,Module) := Module => (M,N) -> (
  cacheModule := youngest(M,N);
  cacheKey := (Ext,M,N);
  if cacheModule#?cacheKey then return cacheModule#cacheKey;
  B := ring M;
  if B != ring N
  then error "expected modules over the same ring";
  if not isCommutative B
  then error "'Ext' not implemented yet for noncommutative rings.";
  if not isHomogeneous B
  then error "'Ext' received modules over an inhomogeneous ring";
  if not isHomogeneous N or not isHomogeneous M
  then error "'Ext' received an inhomogeneous module";
  if N == 0 then B^0
  else if M == 0 then B^0
  else (
    p := presentation B;
    A := ring p;
    I := ideal mingens ideal p;
    n := numgens A;
    c := numgens I;
    if c != codim B
    then error "total Ext available only for complete intersections";
    f := apply(c, i -> I_i);
    pM := lift(presentation M,A);
    pN := lift(presentation N,A);
    M' := cokernel ( pM | p ** id_(target pM) );
    N' := cokernel ( pN | p ** id_(target pN) );
    C := complete resolution M';
    X := local X;
    K := coefficientRing A;
    -- compute the fudge factor for the adjustment of bidegrees
```

```

fudge := if #f > 0 then 1 + max(first \ degree \ f) // 2 else 0;
S := K(monoid [X_1 .. X_c, toSequence A.generatorSymbols,
  Degrees => {
    apply(0 .. c-1, i -> {-2, - first degree f_i}),
    apply(0 .. n-1, j -> { 0, first degree A_j})
  },
  Adjust => v -> {- fudge * v#0 + v#1, - v#0},
  Repair => w -> {- w#1, - fudge * w#1 + w#0}
]);
-- make a monoid whose monomials can be used as indices
Rmon := monoid [X_1 .. X_c, Degrees=>{c:{2}}];
-- make group ring, so 'basis' can enumerate the monomials
R := K Rmon;
-- make a hash table to store the blocks of the matrix
blks := new MutableHashTable;
blks#(exponents 1_Rmon) = C.dd;
scan(0 .. c-1, i ->
  blks#(exponents Rmon_i) = nullhomotopy (- f_i*id_C));
-- a helper function to list the factorizations of a monomial
factorizations := (gamma) -> (
  -- Input: gamma is the list of exponents for a monomial
  -- Return a list of pairs of lists of exponents showing the
  -- possible factorizations of gamma.
  if gamma === {} then { ({}), ({} ) }
  else (
    i := gamma#-1;
    splice apply(factorizations drop(gamma,-1),
      (alpha,beta) -> apply (0..i,
        j -> (append(alpha,j), append(beta,i-j)))));
scan(4 .. length C + 1,
  d -> if even d then (
    scan( exponents \ leadMonomial \ first entries basis(d,R),
      gamma -> (
        s := - sum(factorizations gamma,
          (alpha,beta) -> (
            if blks#?alpha and blks#?beta
            then blks#alpha * blks#beta
            else 0));
        -- compute and save the nonzero nullhomotopies
        if s != 0 then blks#gamma = nullhomotopy s;
        ))));
-- make a free module whose basis elements have the right degrees
spots := C -> sort select(keys C, i -> class i === ZZ);
Cstar := S^(apply(spots C,
  i -> toSequence apply(degrees C_i, d -> {i,first d})));
-- assemble the matrix from its blocks.
-- We omit the sign (-1)^(n+1) which would ordinarily be used,
-- which does not affect the homology.
toS := map(S,A,apply(toList(c .. c+n-1), i -> S_i),
  DegreeMap => prepend_0);
Delta := map(Cstar, Cstar,
  transpose sum(keys blks, m -> S_m * toS sum blks#m),
  Degree => {-1,0});
DeltaBar := Delta ** (toS ** N');
assert isHomogeneous DeltaBar;
assert(DeltaBar * DeltaBar == 0);
-- now compute the total Ext as a single homology module
cacheModule#cacheKey = prune homology(DeltaBar,DeltaBar))

```

Remark 5.5. The bigraded module $\mathrm{Tor}_{\bullet}^B(M, N)$ is the homology of the complex $(C \otimes_A D') \otimes_B N$, where $C \otimes_A D'$ is the complex from Theorem 4.1. Observations parallel to Remarks 4.3 and A.1 show that $\mathrm{Tor}_{\bullet}^B(M, N)$ inherits from D' a structure of bigraded S -module.

It would be desirable also to have algorithms to compute $\mathrm{Tor}_{\bullet}^B(M, N)$ in the spirit of the algorithm presented above for $\mathrm{Ext}_B^{\bullet}(M, N)$. If one of the modules has finite length, then each $\mathrm{Tor}_n^B(M, N)$ is a B -module of finite length, and the computation of $\mathrm{Tor}_{\bullet}^B(M, N)$ can be reduced to a computation of Ext by means of Matlis duality, which here can be realized as vector space duality over the field K . However, in homology there is no equivalent for the finiteness property described in Remark 4.4; it is an **open problem** to devise algorithms that would compute $\mathrm{Tor}_{\bullet}^B(M, N)$ in general.

6 Invariants of Modules

In this section we apply our techniques to develop effective methods for computation (for graded modules over a graded complete intersection) of invariants such as cohomology modules, Poincaré series, Bass series, complexity, critical degree, and support varieties. For each invariant we produce code that computes it, and illustrate the action of the code on some explicit example.

Whenever appropriate, we describe **open problems** on which the computational power of *Macaulay 2* could be unleashed.

Notation 4.7 is used consistently throughout the section.

6.1 Cohomology Modules

We call the bigraded R -module $P = \mathrm{Ext}_B^{\bullet}(M, k)$ the *contravariant cohomology module* of M over B , and the bigraded R -module $I = \mathrm{Ext}_B^{\bullet}(k, M)$ the *covariant cohomology module* of M . Codes that display presentations of the cohomology modules are presented after a detailed discussion of an example.

Example 6.1.1. Let us create the ring $B = K[x, y, z]/(x^3, y^4, z^5)$.

```
i68 : A = K[x,y,z];
i69 : J = trim ideal(x^3,y^4,z^5)
      3  4  5
o69 = ideal (x , y , z )
o69 : Ideal of A
i70 : B = A/J;
```

We trimmed the ideal, so that we know the generators *Macaulay 2* will use.

This time we want a graded B -module M about whose homology we know nothing *a priori*. One way to proceed is to create M as the cokernel of some random matrix of forms; let's try a 3 by 2 matrix of quadratic forms.

```

i71 : f = random (B^3, B^{-2,-3})
o71 = | 27x2+49xy-14y2-23xz-6yz-19z2 38x2y-34xy2+4y3+x2z+16xyz-y2z-5xz ...
      | -5x2+44xy+38y2+40xz+15yz+4z2 -37x2y+51xy2-36y3+26x2z-38xyz-17y ...
      | 21x2-30xy+32y2-47xz+7yz-50z2 -6x2y-14xy2-26y3-7x2z+41xyz+50y2z ...
      3      2
o71 : Matrix B <--- B

```

We can't read the second column of that matrix, so let's display it separately.

```

i72 : f_{1}
o72 = | 38x2y-34xy2+4y3+x2z+16xyz-y2z-5xz2-6yz2+47z3 |
      | -37x2y+51xy2-36y3+26x2z-38xyz-17y2z+17xz2-11yz2+8z3 |
      | -6x2y-14xy2-26y3-7x2z+41xyz+50y2z+26xz2+46yz2-44z3 |
      3      1
o72 : Matrix B <--- B

```

Now let's make the module M .

```
i73 : M = cokernel f;
```

We are going to produce isomorphisms of bigraded modules

$$P^{\text{even}} \cong R[4, 10] \oplus (X_1, X_2)[2, 7] \oplus \left(\frac{R}{(X_1, X_2, X_3)} \right)^3 \oplus R^4[2, 7]$$

$$P^{\text{odd}} \cong \frac{R}{(X_1, X_2, X_3)}[1, 2] \oplus \left(\frac{R}{(X_1)} \right)^3 [3, 9] \oplus \frac{R}{(X_1, X_2)}[1, 3] \oplus R^6[3, 9]$$

over the polynomial ring $R = K[X_1, X_2, X_3]$ over K , bigraded by

$$\text{Deg}(X_1) = (-2, -3) \quad \text{Deg}(X_2) = (-2, -4) \quad \text{Deg}(X_3) = (-2, -5)$$

Let's compute $\text{Ext}_B^*(M, B/(x, y, z))$ by the routine from Section 5.

```

i74 : time P = Ext(M, B^{-1}/(x,y,z));
      -- used 1.64 seconds
i75 : S = ring P;

```

Examine the variables of S ; due to transposing, their bidegrees are displayed with the *opposite* signs.

```

i76 : transpose vars S
o76 = {2, 3} | $X_1 |
      {2, 4} | $X_2 |
      {2, 5} | $X_3 |
      {0, -1} | x |
      {0, -1} | y |
      {0, -1} | z |
      6      1
o76 : Matrix S <--- S

```

The variables x , y , and z of A annihilate P , and so appear in many places in a presentation of P . To reduce the size of such a presentation, we pass to a ring which eliminates those variables.

```

i77 : R = K[X_1..X_3,Degrees => {{-2,-3},{-2,-4},{-2,-5}},
      Adjust => S.Adjust, Repair => S.Repair];

i78 : phi = map(R,S,{X_1,X_2,X_3,0,0,0})

o78 = map(R,S,{X_1, X_2, X_3, 0, 0, 0})
      1 2 3

o78 : RingMap R <--- S

i79 : P = prune (phi ** P);

i80 : transpose vars ring P

o80 = {2, 3} | X_1 |
      {2, 4} | X_2 |
      {2, 5} | X_3 |
      3 1
o80 : Matrix R <--- R

```

As we planned, the original variables x, y, z , which act trivially on the cohomology, are no longer present in the ring. Next we compute presentations

```

i81 : evenPart P

o81 = cokernel {-4, -10} | 0 0 0 0 0 0 0 0 0 0 0 |
      {-4, -10} | 0 0 0 0 0 0 0 0 0 0 0 |
      {-4, -11} | 0 0 0 0 0 0 0 0 0 0 0 |
      {0, 0} | 0 0 X_3 0 0 X_2 0 X_1 0 0 0 |
      {0, 0} | 0 X_3 0 0 X_2 0 X_1 0 0 0 |
      {0, 0} | X_3 0 0 X_2 0 0 0 0 X_1 0 |
      {-2, -7} | 0 0 0 0 0 0 0 0 0 0 0 |
      {-2, -7} | 0 0 0 0 0 0 0 0 0 0 0 |
      {-2, -7} | 0 0 0 0 0 0 0 0 0 0 0 |
      {-2, -7} | 0 0 0 0 0 0 0 0 0 0 0 |
      10
o81 : R-module, quotient of R

i82 : oddPart P

o82 = cokernel {-1, -2} | X_3 0 X_2 0 0 0 0 X_1 |
      {-3, -9} | 0 0 0 0 0 0 X_1 0 |
      {-3, -9} | 0 0 0 0 0 X_1 0 0 |
      {-3, -9} | 0 0 0 0 X_1 0 0 0 |
      {-1, -3} | 0 X_2 0 X_1 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 |
      11
o82 : R-module, quotient of R

```

These presentations yield the desired isomorphisms of bigraded R -modules.

The procedure above can be automated by installing a method that will be run when `Ext` is presented with a module M and the residue field k . It displays a presentation of $\text{Ext}_B^\bullet(M, k)$ as a bigraded R -module.

Code 6.1.2. The function `changeRing H` takes an S -module H and tensors it with R . It does this by constructing R and a ring homomorphism

$$\varphi: A[X_1, \dots, X_c] = S \rightarrow R = K[X_1, \dots, X_c]$$

```
i83 : changeRing = H -> (
      S := ring H;
      K := coefficientRing S;
      degs := select(degrees source vars S,
                    d -> 0 != first d);
      R := K[X_1 .. X_#degs, Degrees => degs,
           Repair => S.Repair, Adjust => S.Adjust];
      phi := map(R,S,join(gens R,(numgens S - numgens R):0));
      prune (phi ** H)
    );
```

Code 6.1.3. The function `Ext(M,k)` computes $\text{Ext}_B^\bullet(M, k)$ when B is a graded complete intersection, M a graded B -module, and k is the residue field of B . The result is presented as a module over the ring $k[X_1, \dots, X_c]$.

```
i84 : Ext(Module, Ring) := (M, k) -> (
      B := ring M;
      if ideal k != ideal vars B
      then error "expected the residue field of the module";
      changeRing Ext(M, coker vars B)
    );
```

Example 6.1.4. For a test, we run again the computation for P^{odd} .

```
i85 : use B;
i86 : k = B/(x,y,z);
i87 : use B;
i88 : P = Ext(M,k);
i89 : time oddPart P
      -- used 0.09 seconds
o89 = cokernel {-1, -2} | X_3 0 X_2 0 0 0 0 X_1 |
      {-3, -9} | 0 0 0 0 0 0 0 X_1 0 |
      {-3, -9} | 0 0 0 0 0 0 X_1 0 0 |
      {-3, -9} | 0 0 0 0 0 X_1 0 0 0 |
      {-1, -3} | 0 X_2 0 X_1 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      {-3, -9} | 0 0 0 0 0 0 0 0 0 |
      ...
o89 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module ...
      1 2 3
```

We also introduce code for computing the covariant cohomology modules.

Code 6.1.5. The function `Ext(k,M)` computes $\text{Ext}_B^\bullet(k, M)$ when B is a graded complete intersection, M a graded B -module, and k is the residue field of B . The result is presented as a module over the ring $k[X_1, \dots, X_c]$.

```

i90 : Ext(Ring,Module) := (k,M) -> (
      B := ring M;
      if ideal k != ideal vars B
      then error "expected the residue field of the module";
      changeRing Ext(coker vars B,M)
    );

```

Let's see the last code in action.

Example 6.1.6. For B and M from Example 6.1.1 we compute the odd part of the covariant cohomology module $\text{Ext}_B^\bullet(k, M)$.

```

i91 : time I = Ext(k,M);
      -- used 14.81 seconds

i92 : evenPart I

o92 = cokernel {0, 6} | 37X_2 37X_1 |
              {0, 6} | -18X_2 -18X_1 |
              {0, 6} | -13X_2 -13X_1 |
              {0, 6} | -37X_2 -37X_1 |
              {0, 6} | 22X_2 22X_1 |
              {0, 6} | 0 0 |
              {0, 6} | X_2 X_1 |
              ...

o92 : K [X_1, X_2, X_3, Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module ...
      1 2 3

i93 : oddPart I

o93 = cokernel {-1, 5} | -48X_3 13X_3 34X_3 3X_3 0 0 0 ...
                    {-1, 5} | 3X_3 -40X_3 8X_3 8X_3 0 0 0 ...
                    {-1, 5} | -X_3 37X_3 -13X_3 -35X_3 0 0 0 ...
                    {-1, 4} | 4X_2 20X_2 3X_2 -47X_2 4X_1 20X_1 3X_1 ...
                    {-1, 4} | 0 51X_2 0 -30X_2 0 51X_1 0 ...
                    {-1, 4} | 0 12X_2 0 -3X_2 0 12X_1 0 ...
                    {-1, 4} | 42X_2 12X_2 46X_2 25X_2 42X_1 12X_1 46X_1 ...
                    {-1, 4} | 45X_2 24X_2 -14X_2 -35X_2 45X_1 24X_1 -14X_1 ...
                    {-1, 4} | 0 0 X_2 0 0 0 X_1 ...
                    {-1, 4} | X_2 0 0 0 X_1 0 0 ...
                    {-1, 4} | 0 -40X_2 0 10X_2 0 -40X_1 0 ...
                    {-1, 4} | 0 X_2 0 0 0 X_1 0 ...
                    {-1, 3} | 0 0 0 X_1 0 0 0 ...
                    ...

o93 : K [X_1, X_2, X_3, Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module ...
      1 2 3

```

6.2 Poincaré Series

The *graded Betti number* of M over B is the number $\beta_{ns}^B(M)$ of direct summands isomorphic to the free module $B[-s]$ in the n 'th module of a minimal free resolution of M over B . It can be computed from the equality

$$\beta_{ns}^B(M) = \dim_K \text{Ext}_B^n(M, k)_s$$

The *graded Poincaré series* of M over B is the generating function

$$P_M^B(t, u) = \sum_{n \in \mathbb{N}, s \in \mathbb{Z}} \beta_{ns}^B(M) t^n u^{-s} \in \mathbb{Z}[u, u^{-1}][[t]]$$

It is easily computable with *Macaulay 2* from the contravariant cohomology module, by using the `hilbertSeries` routine.

Code 6.2.1. The function `poincareSeries2 M` computes the graded Poincaré series of a graded module M over a graded complete intersection B .

First we set up a ring whose elements can serve as Poincaré series.

```
i94 : T = ZZ[t,u,Inverses=>true,MonomialOrder=>RevLex];

i95 : poincareSeries2 = M -> (
    B := ring M;
    k := B/ideal vars B;
    P := Ext(M,k);
    h := hilbertSeries P;
    T' := degreesRing P;
    substitute(h, {T'_0=>t^-1,T'_1=>u^-1})
);
```

The last line in the code above replaces the variables in the Poincaré series provided by the `hilbertSeries` function with the variables in our ring T .

The n th *Betti number* $\beta_n^B(M)$ of M over B is the rank of the n th module in a minimal resolution of M by free B -modules. The *Poincaré series* $P_M^B(t)$ is the generating function of the Betti numbers. There are expressions

$$\beta_n^B(M) = \sum_{s=0}^{\infty} \beta_{ns}^B(M) \quad \text{and} \quad P_M^B(t) = P_M^B(t, 1)$$

Accordingly, the code for $P_M^B(t)$ just replaces in $P_M^B(t, u)$ the variable u by 1.

Code 6.2.2. The function `poincareSeries1 M` computes the Poincaré series of a graded module M over a graded complete intersection B .

```
i96 : poincareSeries1 = M -> (
    substitute(poincareSeries2 M, {u=>1_T})
);
```

Now let's use these codes in computations.

Example 6.2.3. To get a module whose Betti sequence initially decreases, we form an artinian complete intersection B' and take M' to be a cosyzygy in a minimal injective resolution of the residue field k . Since B' is self-injective, this can be achieved by taking a syzygy of k , then transposing its presentation matrix. Of course, we ask *Macaulay 2* to carry out these steps.

```
i97 : A' = K[x,y,z];
i98 : B' = A'/(x^2,y^2,z^3);
```



```

i99 : C' = res(B'^1/(x,y,z), LengthLimit => 6)

      1      3      6      10      15      21      28
o99 = B' <-- B' <-- B' <-- B' <-- B' <-- B' <-- B'

      0      1      2      3      4      5      6

o99 : ChainComplex

i100 : M' = coker transpose C'.dd_5

o100 = cokernel {-5} | -y 0 0 0 z 0 0 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | -x -y 0 0 0 z 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 x -y 0 0 0 z 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 x -y 0 0 0 z 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 -x 0 0 0 0 z 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 y 0 0 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 -x y 0 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 0 x y 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 0 0 x y 0 0 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 0 0 0 -x y 0 0 0 0 0 0 0 0 |
                  {-5} | 0 0 0 0 0 0 0 0 x 0 0 0 0 0 0 0 0 |
                  {-6} | 0 0 0 0 0 0 0 0 0 -y 0 z 0 0 0 0 |
                  {-6} | 0 0 0 0 0 0 0 0 0 0 x -y 0 z 0 0 |
                  {-6} | 0 0 0 0 0 0 0 0 0 0 -x 0 0 z 0 |
                  {-6} | z2 0 0 0 0 0 0 0 0 0 0 y 0 0 0 |
                  {-6} | 0 -z2 0 0 0 0 0 0 0 0 0 x y 0 0 |
                  {-6} | 0 0 z2 0 0 0 0 0 0 0 0 0 -x y 0 |
                  {-6} | 0 0 0 z2 0 0 0 0 0 0 0 0 0 x 0 |
                  {-7} | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 z |
                  {-7} | 0 0 0 0 0 0 0 0 0 z2 0 0 0 0 y |
                  {-7} | 0 0 0 0 0 0 0 0 0 0 z2 0 0 0 x |

```

```

      21
o100 : B'-module, quotient of B'

```

Compute the Poincaré series in two variables $P_{M'}^{B'}(t, u)$.

```

i101 : poincareSeries2 M'

      -7      -6      -5      -6      -5      -4      2 -5      2 - ...
      3u  + 7u  + 11u  + t*u  + 5t*u  + 9t*u  - 6t u  - 14t u  ...
o101 = -----
      ...
      ...

```

```

o101 : Divide

```

Example 6.2.4. We compute $P_M^B(t)$ for the module M from Example 6.1.1.

```

i102 : p = poincareSeries1 M

      2      3      4      5      6      7
      3 + 2t - 5t + 4t + 12t + t - 4t - t
o102 = -----
      2      2      2
      (1 - t)(1 - t)(1 - t)

```

```

o102 : Divide

```

We have written some rather naïve code for simplifying rational functions as above. It locates factors of the form $1 - t^n$ in the denominator, factors out

$1-t$, and factors out $1+t$ if n is even. Keeping the factors of the denominator separate, it then cancels as many of them as it can with the numerator.

```
i103 : load "simplify.m2"

i104 : simplify p

o104 = 
$$\frac{3 - t - 4t^2 + 8t^3 + 4t^4 - 3t^5 - t^6}{(1+t)^2(1-t)^3}$$


o104 : Divide
```

In this case, it succeeded in canceling a factor of $1+t$.

Example 6.2.5. We compute some Betti numbers for M . We use the division operation in the Euclidean domain $T' = \mathbb{Q}[t, t^{-1}]$ with the reverse monomial ordering to compute power series expansions.

```
i105 : T' = QQ[t, Inverses=>true, MonomialOrder=>RevLex];

i106 : expansion = (n,q) -> (
  t := T'_0;
  rho := map(T', T, {t, 1});
  num := rho value numerator q;
  den := rho value denominator q;
  n = n + first degree den;
  n = max(n, first degree num + 1);
  (num + t^n) // den
);
```

Now let's expand the Poincaré series up to t^{20} .

```
i107 : expansion(20,p)

o107 = 
$$3 + 2t + 4t^2 + 10t^3 + 15t^4 + 25t^5 + 32t^6 + 46t^7 + 55t^8 + 73t^9 + \dots$$


o107 : T'
```

Just to make sure, let's compare the first few coefficients with the more pedestrian way of doing the computation, one Ext module at a time.

```
i108 : psi = map(K,B)

o108 = map(K,B,{0, 0, 0})

o108 : RingMap K <--- B

i109 : apply(10, i -> rank (psi ** Ext^i(M, coker vars B)))

o109 = {3, 2, 4, 10, 15, 25, 32, 46, 55, 73}

o109 : List
```

Now we restore t to its original use.

```
i110 : use T;
```

6.3 Complexity

The *complexity* of M is the least $d \in \mathbb{N}$ such that the function

$$n \mapsto \dim_K \operatorname{Ext}_B^n(M, k)$$

is bounded above by a polynomial of degree $d - 1$ (with the convention that the zero polynomial has degree -1). This number, denoted $\operatorname{cx}_B(M)$, was introduced in [2] to measure on a polynomial scale the rate of growth of the Betti numbers of M . It is calibrated so that $\operatorname{cx}_B(M) = 0$ if and only if M has finite projective dimension. Corollary 4.11 yields

$$P_M^B(t) = \frac{p_M^B(t)}{(1-t^2)^c} \quad \text{for some} \quad p_M^B(t) \in \mathbb{Z}[t]$$

Decomposing the right hand side into partial fractions, one sees that $\operatorname{cx}_R(M)$ equals the order of the pole of $P_M^B(t)$ at $t = 1$; in particular, $\operatorname{cx}_R(M, N) \leq c$. However, since we get $P_M^B(t)$ from a computation of the R -module $P = \operatorname{Ext}_B^\bullet(M, k)$, it is natural to obtain $\operatorname{cx}_R(M)$ as the Krull dimension of P .

Code 6.3.1. The function `complexity M` yields the complexity of a graded module M over a graded complete intersection ring B .

```
i111 : complexity = M -> dim Ext(M,coker vars ring M);
```

Example 6.3.2. We compute $\operatorname{cx}_B(M)$ for M from Example 6.1.1.

```
i112 : complexity M
o112 = 3
```

6.4 Critical Degree

The *critical degree* of M is the least integer ℓ for which the minimal resolution F of M admits a chain map $g: F \rightarrow F$ of degree $m < 0$, such that $g_{m+n}: F_{m+n} \rightarrow F_n$ is surjective for all $n > \ell$. This number, introduced in [6] and denoted $\operatorname{crdeg}_B M$, is meaningful over every graded ring B . It is equal to the projective dimension whenever the latter is finite.

When B is a complete intersection it is proved in [6, Sect. 7] that $\operatorname{crdeg}_B M$ is finite and yields important information on the Betti sequence:

- if $\operatorname{cx}_B M \leq 1$, then $\beta_n^B(M) = \beta_{n+1}^B(M)$ for all $n > \operatorname{crdeg}_B M$.
- if $\operatorname{cx}_B M \geq 2$, then $\beta_n^B(M) < \beta_{n+1}^B(M)$ for all $n > \operatorname{crdeg}_B M$.

Thus, it is interesting to know $\operatorname{crdeg}_B M$, or at least to have a good upper bound. Here is what is known, in terms of $h = \operatorname{depth} B - \operatorname{depth}_B M$.

- if $\operatorname{cx}_B M = 0$, then $\operatorname{crdeg}_B M = h$.
- if $\operatorname{cx}_B M = 1$, then $\operatorname{crdeg}_B M \leq h$.
- if $\operatorname{cx}_B M = 2$, then $\operatorname{crdeg}_B M \leq h + 1 + \max\{2\beta_h^B(M) - 1, 2\beta_{h+1}^B(M)\}$.

The first part is the Auslander-Buchsbaum Equality, the second part is proved in [11, Sect. 6], the third is established in [4, Sect. 7].

These upper bounds are realistic: there exist examples in complexity 1 when they are reached, and examples in complexity 2 when they are not more than twice the actual value of the critical degree. If $\text{cx}_R M \geq 3$, then it is an **open problem** whether the critical degree of M can be bounded in terms that do not depend on the action of the cohomology operators.

However, in every concrete case $\text{crdeg}_R M$ can be computed explicitly by using *Macaulay 2*. Indeed, it is proved in [6, Sect. 7] that $\text{crdeg}_R M$ is equal to the highest degree of a non-zero element in the socle of the R -module $\text{Ext}_B^\bullet(M, k)$, that is, the submodule consisting of elements annihilated by (X_1, \dots, X_c) . The socle is naturally isomorphic to $\text{Hom}_B(k, \text{Ext}_B^\bullet(M, k))$, so it can be obtained by standard *Macaulay 2* routines.

For instance, for the module M from Example 6.1.1, we get

```
i113 : k = coker vars ring H;
      i114 : prune Hom(k,H)
      o114 = 0
      i114 : K [X1, X2, x, y, Degrees => {{-2, -2}, {-2, -3}, {0, 1}, {0, ...
```

The degrees displayed above show that $\text{crdeg}_R M = 1$.

Of course, one might prefer to see the number $\text{crdeg}_B M$ directly.

Code 6.4.1. The function `criticalDegree M` computes the critical degree of a graded module M over a graded complete intersection ring B .

```
i115 : criticalDegree = M -> (
      B := ring M;
      k := B / ideal vars B;
      P := Ext(M,k);
      k = coker vars ring P;
      - min ( first \ degrees source gens prune Hom(k,P) )
    );
```

Let's test the new code in a couple of cases.

Example 6.4.2. For the module M of Example 6.1.1 we have

```
i116 : criticalDegree M
      o116 = 1
```

in accordance with what was already observed above.

For the module M' of Example 6.2.3 we obtain

```
i117 : criticalDegree M'
      o117 = 5
```

6.5 Support Variety

Let \overline{K} denote an algebraic closure of K . The *support variety* $V_B^*(M)$ is the algebraic set in \overline{K}^c defined by the annihilator of $\text{Ext}_B^*(M, k)$ over $R = K[X_1, \dots, X_c]$. This ‘geometric image’ of the contravariant cohomology module was introduced in [2] and used to study the minimal free resolution of M . The dimension of the support variety is equal to the complexity $\text{cx}_R(M)$, that we can already compute. There is no need to associate a variety to the covariant cohomology module, see 7.4.

Since $V_B^*(M)$ is defined by homogeneous equations, it is a cone in \overline{K}^c . An important **open problem** is whether every cone in \overline{K}^c that can be defined over K is the variety of some B -module M . By [2, Sect. 6] all linear subspaces and all hypersurfaces arise in this way, but little more is known in general.

Feeding our computation of $\text{Ext}_B^*(M, k)$ to standard *Macaulay 2* routines we write code for determining a set of equations defining $V_B^*(M)$.

Code 6.5.1. The function `supportVarietyIdeal M` yields a set of polynomial equations with coefficients in K , defining the support variety $V_B^*(M)$ in \overline{K}^c for a graded module M over a graded complete intersection B .

```
i118 : supportVarietyIdeal = M -> (
      B := ring M;
      k := B/ideal vars B;
      ann Ext(M,k)
    );
```

As before, we illustrate the code with explicit computations. In view of the open problem mentioned above, we fix a ring and a type of presentation, then change randomly the presentation matrix in the hope of finding an ‘interesting’ variety. The result of the experiment is assessed in Remark 6.5.3.

Example 6.5.2. Let \mathbb{F}_7 denote the prime field with 7 elements, and form the zero-dimensional complete intersection $B'' = \mathbb{F}_7[x, y, z]/(x^7, y^7, z^7)$.

```
i119 : K'' = ZZ/7;
i120 : A'' = K''[x,y,z];
i121 : J'' = ideal(x^7,y^7,z^7);
o121 : Ideal of A''
i122 : B'' = A''/J'';
```

We apply the code above to search, randomly, for some varieties. Using `scan` we print the results from several runs with one command.

```
i123 : scan((1,1) .. (3,3), (r,d) -> (
      V := cokernel random (B''^r, B''^{-d});
      << "----- ...
      << endl
      << "V = " << V << endl
      << "support variety ideal = "
      << timing supportVarietyIdeal V
      << endl))
```

```

-----
V = cokernel | -2x+3y+2z |
support variety ideal = ideal (X2 - 2X3, X1 + X3)
-- 0.7 seconds
-----
V = cokernel | 3x2-2xy+xz-3yz |
support variety ideal = ideal(X1 + 3X2 + 2X3)
-- 0.48 seconds
-----
V = cokernel | -2x3+3x2y+y3-x2z-3y2z-xz2-3z3 |
support variety ideal = 0
-- 1.54 seconds
-----
V = cokernel | -3y+3z |
| -2x-2y |
support variety ideal = ideal(X1 + X2 - X3)
-- 0.86 seconds
-----
V = cokernel | -x2+2y2-xz+yz+3z2 |
| 2xy-3xz-3yz-2z2 |
support variety ideal = 0
-- 1.31 seconds
-----
V = cokernel | -x3-2x2y-xy2-2xyz+3y2z+2xz2-yz2-2z3 |
| 2xy2+3y3-3x2z-2y2z+2xz2+2yz2 |
support variety ideal = 0
-- 2.21 seconds
-----
V = cokernel | 3x-y-z |
| -3x-y+2z |
| x-2y+3z |
support variety ideal = 0
-- 1.1 seconds
-----
V = cokernel | 2x2-2xy+2y2+2xz-3z2 |
| -x2+2xy+y2+3xz+3yz-z2 |
| -2xz+2yz+2z2 |
support variety ideal = 0
-- 1.67 seconds
-----
V = cokernel | 2x3-x2y+2xy2-y3-2xyz+3y2z+xz2+3yz2+z3 |
| -3x3-3x2y+3xy2+2x2z+3xyz-3y2z-xz2 |
| -3x3-2x2y-xy2-2y3-2xyz+y2z+xz2+3yz2-z3 |
support variety ideal = 0
-- 1.92 seconds

```

Remark 6.5.3. The (admittedly short) search above did not turn up any non-linear variety. This should be contrasted with the known result that *every* cone in $\overline{\mathbb{F}}_7^3$ is the support variety of some B'' -module.

Indeed, B'' is isomorphic to the group algebra $\mathbb{F}_7[G]$ of the elementary abelian group $G = \mathbb{C}_7 \times \mathbb{C}_7 \times \mathbb{C}_7$, where \mathbb{C}_7 is a cyclic group of order 7. It is shown in [2, Sect. 7] that $V_{B''}^*(V)$ is equal to a variety $V_G^*(V)$, defined in a different way in [9] by Carlson. He proves in [10] that if K is a field of

characteristic $p > 0$, and G is an elementary abelian p -group of rank c , then every cone in \overline{K}^c is the rank variety of a finitely generated module over $K[G]$.

6.6 Bass Series

The *graded Bass number* $\mu_B^{ns}(M)$ of M over B is the number of direct summands isomorphic to $U[s]$ in the n 'th module of a minimal graded injective resolution of M over B , where U is the injective envelope of k . It satisfies

$$\mu_B^{ns}(M) = \dim_K \operatorname{Ext}_B^n(k, M)_s$$

The *graded Bass series* of M over B is the generating function

$$I_B^M(t, u) = \sum_{n \in \mathbb{N}, s \in \mathbb{Z}} \mu_B^{ns}(M) t^n u^s \in \mathbb{Z}[u, u^{-1}][[t]]$$

It is easily computable with *Macaulay 2* from the covariant cohomology module, by using the `hilbertSeries` routine.

Code 6.6.1. The function `bassSeries2 M` computes the graded Bass series of a graded module M over a graded complete intersection B .

```
i124 : bassSeries2 = M -> (
      B := ring M;
      k := B/ideal vars B;
      I := Ext(k, M);
      h := hilbertSeries I;
      T' := degreesRing I;
      substitute(h, {T'_0=>t^-1, T'_1=>u})
    );
```

As with Betti numbers and Poincaré series, there are ungraded versions of Bass numbers and Bass series; they are given, respectively, by

$$\mu_B^n(M) = \sum_{s=0}^{\infty} \mu_B^{ns}(M) \quad \text{and} \quad I_B^M(t) = I_B^M(t, 1)$$

Code 6.6.2. The function `bassSeries1 M` computes the Bass series of a graded module M over a graded complete intersection B .

```
i125 : bassSeries1 = M -> (
      substitute(bassSeries2 M, {u=>1_T})
    );
```

Now let's use these codes in computations.

Example 6.6.3. For k , the residue field of B , the contravariant and covariant cohomology modules coincide. For comparison, we compute side by side the Poincaré series and the Bass series of k , when $B = K[x, y, z]/(x^3, y^4, z^5)$ is the ring defined in Example 6.1.1.

```

i126 : use B;
i127 : L = B^1/(x,y,z);
i128 : p = poincareSeries2 L
o128 = 
$$\frac{1 + 3t^2u + 3t^2u^2 + t^3u^3}{(1 - t^2u)(1 - t^2u^2)(1 - t^2u^3)}$$

i128 : Divide
i129 : b = bassSeries2 L
o129 = 
$$\frac{1 + 3t^{-1}u + 3t^{-2}u^2 + t^{-3}u^3}{(1 - t^{-2}u)(1 - t^{-2}u^2)(1 - t^{-2}u^3)}$$

o129 : Divide

```

The reader would have noticed that the two series are different, and that one is obtained from the other by the substitution $u \mapsto u^{-1}$. This underscores the different meanings of the graded Betti numbers and Bass numbers.

Example 6.6.4. Here we compute the graded and ungraded Bass series of the B -module M of Example 6.1.1.

```

i130 : b2 = bassSeries2 M
o130 = 
$$\frac{7u^6 + t^3u^3 + 9t^4u^4 + 3t^5u^5 - t^2u^2 - t^2u^3 - 4t^3u^3 - 3t^3u^3 - 3t^3u^2 + \cdots}{(1 - t^2u)(1 - t^2u^2)(1 - t^2u^3)}$$

o130 : Divide
i131 : b1 = bassSeries1 M;
i132 : simplify b1
o132 = 
$$\frac{7 + 6t^2 - 8t^3 - 2t^4 + 3t^4}{(1 + t)^2(1 - t)^3}$$

o132 : Divide

```

7 Invariants of Pairs of Modules

In this final section we compute invariants of a pair (M, N) of graded modules over a graded complete intersection B , derived from the reduced Ext module $\text{ext}_B^*(M, N)$ defined in Remark 4.10. The treatment here is parallel to that in Section 6. When one of the modules M or N is equal to the residue field k , the invariants discussed below reduce to those treated in that section.

7.1 Reduced Ext Module

The reduced Ext module $\text{ext}_B^\bullet(M, N) = \text{Ext}_B^\bullet(M, N) \otimes_A k$ defined in Remark 4.10 is computed from our basic routine `Ext(M,N)` by applying the function `changeRing` defined in Code 6.1.2.

Code 7.1.1. The function `ext(M,N)` computes $\text{ext}_B^\bullet(M, N)$ when M and N are graded modules over a graded complete intersection B .

```
i133 : ext = (M,N) -> changeRing Ext(M,N);
```

Example 7.1.2. Using the ring $B = K[x, y, z]/(x^3, y^4, z^5)$ and the module M created in Example 6.1.1, we make new modules

$$N = B/(x^2 + z^2, y^3) \quad \text{and} \quad N' = B/(x^2 + z^2, y^3 - 2z^3)$$

```
i134 : use B;
i135 : N = B^1/(x^2 + z^2, y^3);
i136 : time rH = ext(M,N);
      -- used 15.91 seconds
i137 : evenPart rH
o137 = cokernel {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 0 0 0 0 0 0 X_3 0 0 0 ...
      {0, 2} | 0 0 0 0 0 0 X_3 0 0 0 0 0 ...
      {0, 2} | 0 0 0 0 0 0 0 X_3 0 0 0 0 ...
      {0, 2} | 0 0 0 0 X_3 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 X_3 0 0 0 0 0 0 0 0 0 ...
      {0, 2} | 0 X_3 0 0 0 0 0 0 0 0 0 X_2 ...
      {0, 2} | X_3 0 0 0 0 0 0 0 0 0 X_2 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {0, 1} | 0 0 0 0 0 0 0 0 0 X_2 0 0 ...
      ...
o137 : K [X , X , X , Degrees => {-2, -3}, {-2, -4}, {-2, -5}]-modul ...
      1 2 3
      ...
i138 : oddPart rH
o138 = cokernel {-3, -6} | 0 0 0 0 0 0 0 0 0 0 0 ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 0 0 ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 0 0 ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 0 0 ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 0 0 ...
      {-1, -1} | -39X_3 0 0 0 0 0 0 0 0 X_2 0 ...
      {-1, -1} | 31X_3 0 0 0 0 0 0 0 X_2 0 0 ...
      {-1, -1} | -34X_3 0 0 0 0 0 X_2 0 0 0 0 ...
      {-1, -1} | -35X_3 0 0 0 0 X_2 0 0 0 0 ...
```

```

      {-1, -1} | -29X_3 0 0 0 X_2 0 0 0 0 0 0 ...
      {-1, -1} | 12X_3 0 0 X_2 0 0 0 0 0 0 0 ...
      {-1, -1} | -8X_3 0 X_2 0 0 0 0 0 0 0 0 ...
      {-1, -1} | X_3 X_2 0 0 0 0 0 0 0 0 0 ...
      {-3, -7} | 0 0 0 0 0 0 0 0 0 0 X_1 ...
      ...
o138 : K [X , X , X , Degrees => {-2, -3}, {-2, -4}, {-2, -5}]-modul ...
      1 2 3
      ...

i139 : N' = B^1/(x^2 + z^2,y^3 - 2*z^3);

i140 : time rH' = ext(M,N');
      -- used 20.26 seconds

i141 : evenPart rH'

o141 = cokernel {-4, -8} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -8} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 0 0 0 0 0 X_3 0 0 0 0 ...
      {0, 2} | 0 0 0 0 0 0 X_3 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 0 0 X_3 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 X_3 0 0 0 0 0 0 0 0 X_2 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 0 ...
      {0, 2} | 0 0 0 X_3 0 0 0 0 0 0 0 0 ...
      {0, 1} | 0 0 0 0 0 0 0 0 X_2 0 0 0 ...
      {0, 1} | 0 0 0 0 0 0 0 0 0 X_2 0 0 ...
      ...

o141 : K [X , X , X , Degrees => {-2, -3}, {-2, -4}, {-2, -5}]-modul ...
      1 2 3
      ...

i142 : oddPart rH'

o142 = cokernel {-3, -6} | 0 0 0 0 0 0 0 0 -42X_2 21X_ ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 -6X_2 -32X ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 -8X_2 12X_ ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 26X_2 -36X ...
      {-3, -6} | 0 0 0 0 0 0 0 0 0 50X_2 18X_ ...

```

```

{-3, -6} | 0 0 0 0 0 0 0 0 0 31X_2 7X_2 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 0 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 0 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 0 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 0 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 X_1 ...
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 X_1 ...
{-1, -2} | 0 0 0 X_2 0 0 0 0 X_1 0 0 ...
{-1, -2} | 0 0 X_2 0 0 0 0 X_1 0 0 ...
{-1, -2} | 0 X_2 0 0 0 0 X_1 0 0 0 ...
{-1, -2} | X_2 0 0 0 X_1 0 0 0 0 0 ...
...
o142 : K [X , X , X , Degrees => {-2, -3}, {-2, -4}, {-2, -5}]-modul ...
      1 2 3 ...

```

7.2 Ext-generator Series

The Ext-generator series $G_B^{M,N}(t, u)$ defined in Remark 4.10 generalizes both the Poincaré series of M and the Bass series of N , as seen from the formulas

$$P_M^B(t, u) = G_B^{M,k}(t, u) \quad \text{and} \quad I_B^N(t, u) = G_B^{k,N}(t, u^{-1})$$

Similar equalities hold for the corresponding series in one variable. Codes for computing Ext-generator series are easy to produce.

Code 7.2.1. The function `extgenSeries2(M,N)` computes $G_B^{M,N}(t, u)$ when M and N are graded modules over a graded complete intersection B , and presents it as a rational function with denominator $(1 - t^2 u^{r_1}) \cdots (1 - t^2 u^{r_c})$.

```

i143 : extgenSeries2 = (M,N) -> (
      H := ext(M,N);
      h := hilbertSeries H;
      T' := degreesRing H;
      substitute(h, {T'_0=>t^-1,T'_1=>u^-1})
    );

```

Code 7.2.2. The function `extgenSeries1(M,N)` computes the Ext-generator series in one variable for a pair (M, N) of graded modules over a graded complete intersection B .

```

i144 : extgenSeries1 = (M,N) -> (
      substitute(extgenSeries2(M,N), {u=>1_T})
    );

```

Example 7.2.3. For M, N , and N' as in Example 7.1.2 we obtain

```

i145 : time extgenSeries2(M,N)
      -- used 0.44 seconds
      -2  -1      2      2 2      2 3      2 4      3 4      3 ...
      8u  + u  + 8t*u - 8t u - 9t u - 9t u + 7t u - 8t u - 8t u ...
o145 = ----- ...
      ...
o145 : Divide

```

```

i146 : g=time extgenSeries1(M,N)
-- used 0.13 seconds

          2      3      4      5      6      7
      9 + 8t - 19t - 11t + 17t + 4t - 7t - t
o146 = -----
          2      2      2
      (1 - t)(1 - t)(1 - t)

o146 : Divide

i147 : simplify g

          2      3      4
      9 - t - 9t + 6t + t
o147 = -----
          2
      (1 + t)(1 - t)

o147 : Divide

i148 : time extgenSeries2(M,N')
-- used 0.15 seconds

      -2      -1      2      2      2 2      2 3      2 4      3 5      ...
      7u + 2u + 4t*u - 7t u - 9t u - 9t u + 16t u - 4t u + 2 ...
o148 = -----
      ...
      ...

o148 : Divide

i149 : g'=time extgenSeries1(M,N')
-- used 0.18 seconds

          2      3      4      5      6
      9 + 4t - 9t + 4t + 8t - 2t - 2t
o149 = -----
          2      2      2
      (1 - t)(1 - t)(1 - t)

o149 : Divide

i150 : simplify g'

          2      3      5
      9 - 5t - 4t + 8t - 2t
o150 = -----
          2      3
      (1 + t)(1 - t)

o150 : Divide

```

7.3 Complexity

The *complexity* of a pair of B -modules (M, N) is the least $d \in \mathbb{N}$ such that there exists a polynomial of degree $d - 1$ bounding above the function

$$n \mapsto \dim_K \operatorname{ext}_B^n(M, N)$$

It is denoted $\text{cx}_B(M, N)$ and measures on a polynomial scale the rate of growth of the minimal number of generators of $\text{Ext}_B^n(M, N)$; it vanishes if and only if $\text{Ext}_B^n(M, N) = 0$ for all $n \gg 0$. Corollary 4.11 yields

$$G_B^{M,N}(t) = \frac{h(t)}{(1-t^2)^c} \quad \text{for some } h(t) \in \mathbb{Z}[t]$$

so decomposition into partial fractions shows that $\text{cx}_R(M, N)$ equals the order of the pole of $G_B^{M,N}(t)$ at $t = 1$. Alternatively, $\text{cx}_R(M, N)$ can be obtained by computing the Krull dimension of a reduced Ext module over R .

Code 7.3.1. The function `complexityPair(M,N)` yields the complexity of a pair (M, N) of graded modules over a graded complete intersection ring B .

```
i151 : complexityPair = (M,N) -> dim ext(M,N);
```

Example 7.3.2. For M , N , and N' as in Example 7.1.2 we have

```
i152 : time complexityPair(M,N)
-- used 0.39 seconds

o152 = 2

i153 : time complexityPair(M,N')
-- used 0.12 seconds

o153 = 3
```

7.4 Support Variety

Let \overline{K} be an algebraic closure of K . The *support variety* $V_B^*(M, N)$ is the algebraic set in \overline{K}^c defined by the annihilator of $\text{ext}_B^\bullet(M, N)$ over $R = K[X_1, \dots, X_c]$. It is clear from the definition that $V_B^*(M, k)$ is equal to the variety $V_B^*(M)$ defined in 6.5. One of the main results of [5, Sect. 5] shows that $V_B^*(M, N) = V_B^*(M) \cap V_B^*(N)$, and, as a consequence, $V_B^*(M) = V_B^*(M, M) = V_B^*(k, M)$. The dimension of $V_B^*(M, N)$ is equal to the complexity $\text{cx}_R(M, N)$, already computed above.

Feeding our computation of $\text{ext}_B^\bullet(M, N)$ to standard *Macaulay 2* routines we write code for determining a set of equations defining $V_B^*(M, N)$.

Code 7.4.1. The function `supportVarietyPairIdeal(M,N)` yields a set of polynomial equations with coefficients in K , defining the variety $V_B^*(M, N)$ in \overline{K}^c for graded modules M, N over a graded complete intersection B .

```
i154 : supportVarietyPairIdeal = (M,N) -> ann ext(M,N);
```

Example 7.4.2. For M , N , and N' as in Example 7.1.2 we have

```
i155 : time supportVarietyPairIdeal(M,N)
-- used 0.97 seconds

o155 = ideal X
      1

o155 : Ideal of K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]
      1  2  3
```

```

i156 : time supportVarietyPairIdeal(M,N')
-- used 1.73 seconds

o156 = 0

o156 : Ideal of K [X1, X2, X3, Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]

```

Appendix A. Gradings

Our purpose here is to set up a context in which the theory of Sections 3 and 4 translates into data that *Macaulay 2* can operate with.

A first point is to develop a *flexible* and *consistent* scheme within which to handle the two kinds of degrees we deal with—the internal gradings of the input, and the homological degrees created during computations.

A purely technological difficulty arises when our data are presented to *Macaulay 2*: it only accepts multidegrees whose first component is positive, which is *not* the case for rings of cohomology operators.

A final point, mostly notational, tends to generate misunderstanding and errors if left unaddressed. On the printed page, the difference between homological and cohomological conventions is handled graphically by switching between sub- and super-indices, and reversing signs; both authors were used to it, but *Macaulay 2* has so far refused to read \TeX printouts.

The *raison d'être* of the following remarks was to debug communications between the three of us.

Remark A.1. Only one degree, denoted deg , appears in Section 2, and anywhere in the main text before Notation 4.7; when needed, it will be referred to as *homological degree*.

Assume that $A = \bigoplus_{h \in \mathbb{Z}} A_h$ is a graded ring. Any element a of A_h is said to be homogeneous of *internal degree* h ; the notation for this is $\text{deg}' a = h$. Let $\mathbf{f} = \{f_1, \dots, f_c\}$ be a Koszul-regular set consisting of homogeneous elements. We give the ring $B = A/(\mathbf{f})$ the induced grading, and extend the notation for internal degree to all graded B -modules M .

Let M be a graded B -module. For any integer e , we let $M[e]$ denote the graded module with $M[e]_d = M_{d+e}$. We take a projective resolution C of M by graded A -modules, with differential d_C preserving internal degrees. Recall that we have been writing $\text{deg } x = n$ to indicate that x is an element in C_n ; we refer to this situation also by saying that x has *homological degree* x . We combine both degrees in a single *bidegree*, denoted Deg , as follows:

$$\text{Deg } x = (\text{deg } x, \text{deg}' x)$$

For a bigraded module H and pair of integers (e, e') , we let $H[e, e']$ denote the bigraded module with $H[e, e']_{d, d'} = H_{d+e, d'+e'}$.

Because $\text{deg } Y_i = 2$, the elements of the free B -module Q have homological degree 2. We introduce an internal grading deg' on Q by setting $\text{deg}' Y_i = r_i$,

where $r_i = \deg' f_i$ for $i = 1, \dots, c$. With this choice, the homomorphism $f: Q \rightarrow A$ acquires internal degree 0 (of course, this was the reason behind our choice of grading in the first place). The internal grading on Q defines, in the usual way, internal gradings on all symmetric and exterior powers of Q and Q^* ; in particular, $\deg' Y^{(\alpha)} = \sum \alpha_i r_i$ and $\deg' Y^{\wedge \beta} = \sum \beta_i r_i$. Thus, the ring $S = A[X_1, \dots, X_c]$ acquires a bigrading defined by $\text{Deg } a = (0, h)$ for all elements $a \in A_h$ and $\text{Deg } X_i = (-2, -r_i)$ for $i = 1, \dots, c$.

In this context, we call S the *bigraded ring of cohomology operators*.

Since the differential d_C has internal degree 0, a null-homotopic chain map $C \rightarrow C$ which is homogeneous of internal degree r will have a null-homotopy that is itself homogeneous of internal degree r . In the proof of Theorem 3.1 we construct maps d_γ as null-homotopies, so we may arrange for them to be homogeneous maps with $\deg' d_\gamma = \sum \gamma_i d_i$. Our grading assumptions guarantee that d is homogeneous with $\text{Deg } d = (-1, 0)$.

With these data, the B -free resolution $C \otimes_A D'$ provided by Theorem 4.1 becomes one by graded B -modules, and its differential ∂ is homogeneous with $\text{Deg } \partial = (-1, 0)$. For any graded B -module N , these properties are transferred to the complex $\text{Hom}_B(C \otimes_A D', N)$ and its differential.

We sum up the contents of Remarks 4.3 and A.1.

Remark A.2. If A is a graded ring, $\{f_1, \dots, f_c\}$ is a Koszul-regular set consisting of homogeneous elements, B is the residue ring $A/(\mathbf{f})$, and M, N are graded B -modules, then $\text{Ext}_B^\bullet(M, N)$ is a bigraded module over the ring $S = A[X_1, \dots, X_c]$, itself bigraded by setting $\text{Deg } a = (0, \deg'(a))$ for all homogeneous $a \in A$ and $\text{Deg } X_i = (-2, -\deg'(f_i))$ for $i = 1, \dots, c$.

Remark A.3. The core algorithms of the program can handle multi-graded rings and modules, but only if each variable in the ring has positive first component of its multi-degree. At the moment, a user who needs a multi-graded ring R which violates this requirement must provide two linear maps: **R.Adjust**, that transforms the desired multi-degrees into ones satisfying this requirement, as well as its inverse map, **R.Repair**. The routine **Ext**, discussed above, incorporates such adjustments for the rings of cohomology operators it creates. When we wish to create related rings with some of the same multi-degrees, we may use the same adjustment operator.

References

1. David J. Anick: A counterexample to a conjecture of Serre. *Ann. of Math. (2)*, 115(1):1–33, 1982 and 116(3):661, 1982.
2. Luchezar L. Avramov: Modules of finite virtual projective dimension. *Invent. Math.*, 96(1):71–101, 1989.
3. Luchezar L. Avramov: Infinite free resolutions. In *Six lectures on commutative algebra (Bellaterra, 1996)*, pages 1–118. Progress in Math., vol 166, Birkhäuser, Basel, 1998.

4. Luchezar L. Avramov and Ragnar-Olaf Buchweitz: Homological algebra modulo a regular sequence with special attention to codimension two. *J. Algebra*, 230(1):24–67, 2000.
5. Luchezar L. Avramov and Ragnar-Olaf Buchweitz: Support varieties and cohomology over complete intersections. *Invent. Math.*, 142(2):285–318, 2000.
6. Luchezar L. Avramov, Vesselin N. Gasharov, and Irena V. Peeva: Complete intersection dimension. *Inst. Hautes Études Sci. Publ. Math.*, 86:67–114, 1997.
7. Luchezar L. Avramov and Irena Peeva: Finite regularity and Koszul algebras. *Amer. J. Math*, 123(to appear), 2001.
8. Luchezar L. Avramov and Li-Chuan Sun: Cohomology operators defined by a deformation. *J. Algebra*, 204(2):684–710, 1998.
9. Jon F. Carlson: The varieties and the cohomology ring of a module. *J. Algebra*, 85(1):104–143, 1983.
10. Jon F. Carlson: The variety of an indecomposable module is connected. *Invent. Math.*, 77(2):291–299, 1984.
11. David Eisenbud: Homological algebra on a complete intersection, with an application to group representations. *Trans. Amer. Math. Soc.*, 260(1):35–64, 1980.
12. Tor H. Gulliksen: A change of ring theorem with applications to Poincaré series and intersection multiplicity. *Math. Scand.*, 34(1):167–183, 1974.
13. Vikram Mehta: *Endomorphisms of complexes and modules over Golod rings*. Thesis. University of California, Berkeley, CA, 1975.
14. Jan-Erik Roos: Commutative non-Koszul algebras having a linear resolution of arbitrarily high order. Applications to torsion in loop space homology. *C. R. Acad. Sci. Paris Sér. I Math.*, 316(11):1123–1128, 1993.
15. Jack Shamash: The Poincaré series of a local ring. *J. Algebra*, 12:453–470, 1969.

Index

- assert** 7
- Betti number 32
- bidegree 46
- bigraded ring of cohomology operators 47
- bounded filtration 12
- Cartan map 10
- code** 25
- complete intersection 1
- complexity 35, 44
- contravariant cohomology module 27
- covariant cohomology module 27
- critical degree 35
- dd** 4
- divided powers algebra 9
- Ext** 1, 25
- Ext-generator series 19
- exterior algebra 9
- generates 5
- graded Bass number 39
- graded Bass series 39
- graded Betti number 31
- graded complete intersection 1
- graded Poincaré series 32
- hilbertSeries** 32, 39
- homogeneous derivation 9
- homological degree 46
- homology** 21
- internal degree 46
- Koszul map 10
- Koszul-regular 16
- Leibniz rule 9
- matrix factorization 3
- nullhomotopy** 4
- Poincaré series 32
- reduced Ext module 19
- reduced ring of cohomology operators 19
- resolution** 20
- ring of cohomology operators 17
- scan** 37
- support variety 37, 45
- symmetric algebra 9